

前所未見的 EXCEL 寶典

{點、陣列、VBA 運算}

egg

January 15, 2018

目錄

II 二維	1
4 陣列	3
4.1 何謂陣列	3
4.2 陣列輸入	4
4.2.1 陣列輸入—列	6
4.2.2 陣列輸入—欄	9
4.2.3 陣列輸入—列欄	10
5 基本陣列運算	13
5.1 數學上矩陣運算	13
5.1.1 矩陣運算—加法	13
5.1.2 矩陣運算—乘法	14
5.2 Excel 陣列運算	15
5.2.1 Excel 陣列運算—加法	15
5.2.2 Excel 陣列運算—乘法	17
5.3 基本綜合陣列運算	18
5.3.1 多重結果	18
5.3.2 單一結果	21
6 陣列函數運算	25
7 {=SUM(IF())}	27
7.1 資料設定	27
7.2 問題設定	30
7.3 前提	31

7.4	$\overbrace{\text{CSE}}$	34
7.4.1	多少男性消費	34
7.4.2	男性消費金額	37
7.4.3	已婚男性消費金額	38
7.4.4	已婚男性一年內消費金額	39
7.4.5	49 歲已婚男性一年內消費金額	40
7.4.6	20~30 歲已婚男性一年內消費金額	41
7.4.7	20~30 及 40~50 歲男性一年內消費金額	42
7.4.8	20~30 歲女性及 40~50 歲男性一年內消費金額	43
7.4.9	20~30 歲女性及 40~50 歲未婚男性一年內消費金額	44
7.4.10	{=SUM(IF(IF(...)))}	46
7.5	只有 SUM 能 IF 嗎?	50
7.6	與 =SUMIFS() 比較	52
7.6.1	=SUMIFS()	53
7.6.2	資料設定	53
7.6.3	男性消費金額	55
7.6.4	已婚男性消費金額	56
7.6.5	已婚男性一年內消費金額	56
7.6.6	49 歲已婚男性一年內消費金額	57
7.6.7	20~30 歲已婚男性一年內消費金額	57
7.6.8	20~30 及 40~50 歲男性一年內消費金額	58
7.6.9	20~30 歲女性及 40~50 歲男性一年內消費金額	59
7.6.10	20~30 歲女性及 40~50 歲未婚男性一年內消費金額	60
7.6.11	{=SUM(IF())} 與 =SUMIFS() 比較	62
7.7	表格	62
8	陣列運算的運用	63
8.1	用函數排序	63
8.2	用函數篩選	64
8.2.1	=SMALL()	65
8.2.2	=INDEX()	66
8.2.3	=IFERROR()	67
8.3	抽樣—抽出不放回	68

8.3.1	{=ROW()}	70
8.3.2	{=COUNTIF()}	71
8.3.3	{=IF(COUNTIF())}	73
8.3.4	{=LARGE(IF(COUNTIF()))}	74
9	例子演練	75
9.1	半年內消費筆數	75

Part II

二維


第 4 章

陣列


我們先介紹數學上矩陣與 Excel 陣列的概念, 陣列輸入及基本運算, 接著進入陣列運算的運用之前, 我們先介紹一下常用的函數, 這些函數結合陣列的運算, 如基本款 $\{=SUM(IF)\}$, 功能會強大到讓人驚訝, 舉一些實例來理解運作方式, 再結合深一點的函數可以更無限寬廣, 最後舉一些例子來運用。

4.1 何謂陣列

介紹陣列之前, 先要講一下欄、列。表格中垂直方向是欄 (column), 欄的計數名稱是以英文字母 (A, B, C,...) 來命名,




水平方向還是列 (row), 列的計數名稱是以阿拉伯數字 (1, 2, 3,...) 來命名,



由「欄名列號」來表示 Excel 中的位址, 第一欄跟第一列交集所指的就是 A1 位址的儲存格。

再來，陣列其實就是一個範圍的儲存格，範圍包含 m 個列與 n 個欄，所以陣列就是一個 $m \times n$ 的範圍，例如 3 個欄與 4 個列的區域組成一個 4×3 陣列。如下圖，B2:D5 的範圍就是一個 (4×3) 陣列。



The image shows an Excel spreadsheet with columns labeled A, B, C, D and rows numbered 1 to 8. The data is as follows:

	A	B	C	D
1	姓名	國文	英文	數學
2	張三	81	58	33
3	李四	92	51	66
4	王五	88	70	72
5	周六	79	64	58
6				
7				
8				

A red rectangle highlights the range B2:D5, which contains the scores for students 張三, 李四, 王五, and 周六. The value 81 in cell B2 is also highlighted with a black box.

4.2 陣列輸入

在一般的公式中，您可以在包含值的儲存格中輸入參照，或輸入稱為常數的值。同樣地，您也可以在陣列中輸入參照，或是輸入儲存格中包含的值的陣列（也稱為常數陣列）。陣列公式可使用與非陣列公式相同的方式來接受常數，但是您必須按特定格式輸入常數陣列。

常數陣列可以包含數字、文字、邏輯值（例如 TRUE 或 FALSE）或錯誤值（例如 #N/A）。不同類型的值可以在相同的常數陣列中，例如 {1, 3, 4; TRUE, FALSE, TRUE}。常數陣列中的數字可以是整數、小數或科學記號格式。文字前後必須括上雙引號，例如“星期二”。


常數陣列中不得包含儲存格參照、長度不等的欄或列、公式或是特殊字元 \$（貨幣符號）、括號或 %（百分比符號）。

設定常數陣列格式時，請確認：


- 使用大括號 ({ }) 括住常數。
- 使用逗號 (,) 區隔不同欄的值。例如，輸入 {10, 20, 30, 40} 代表 10、20、30 和 40 等值。這種常數陣列稱作 1 對 4 陣列，相當於 1 列 4 欄的參照。
- 使用分號 (;) 區隔不同列的值。例如，若要表示一列中的四個值 10、20、30、40 以及下一列中的 50、60、70 和 80 四個值，您必須輸入一個 2 對 4 常數陣列：{10, 20, 30, 40; 50, 60, 70, 80}。

以上是 Microsoft 對常數陣列的解說。

單個儲存格輸入時，只要選取 A1 格，鍵入 1，再按 **Enter** 鍵，Excel 的 A1 格就會顯示“1”，資料編輯列也顯示為“1”，如圖，




當然你也可以輸入“=1”，得出一樣的結果，只是在資料編輯列多一個等號。



當要輸入陣列（一欄、一行或多欄多列）時，在資料編輯列輸入完內容後，再也不是單單鍵入 **Enter** 鍵，而是要同時按 **Ctrl** + **Shift** + **Enter** 三鍵，而 Excel 會用大括號 {} 將運算式括起來，這才算是完成陣列輸入。

單格的輸入通常都是按 **Enter** 鍵，當然也可以透過陣列輸入的方式，按 **Ctrl** + **Shift** + **Enter** 三鍵來完成輸入，也就把單格看成是 1 × 1 的陣列就行了，以下我們看 2 個例子。



A1 儲存格手動輸入“=1”，B1 儲存格手動輸入“={1}”，最後按 **Ctrl** + **Shift** + **Enter** 三鍵來完成輸入，得到相同的結果，都為 1。至於最外層的大括號，這是我們要介紹的陣列輸入，往下看就會了解。

另外還有詳細的陣列輸入的方法及規定，讓我們來了解陣列的輸入吧。

CSE


[A]

Ctrl + Shift ↑ + Enter

以下我們把 Ctrl + Shift ↑ + Enter 程序簡稱叫做 CSE, CSE 不但是陣列輸入必要程序, 也是陣列運算必要程序, 接著我們開始一一介紹輸入及運算的 CSE。

4.2.1 陣列輸入—列

假設我們要在—列 1×5 的儲存格中分別輸入 1~5, 譬如在 A1 到 E1 的 5 個儲存格。




	A	B	C	D	E	F
1	1	2	3	4	5	
2						

最簡單的想法應該是一格一格輸入, 但這裡我們要介紹的陣列輸入法。

用陣列輸入法時,

1. 選取儲存格 A1 到 E1
2. 鍵入 “={”
3. 在大括弧中輸入 “1, 2, 3, 4, 5”
4. 鍵入 CSE

圖解如下,



步驟 1: 因為陣列輸入, 所以一次選取所要輸入的所有儲存格 A1 到 E1。

步驟 2: 先輸入等號, 再輸入大括號, 建議每遇到括號 (不論大、中、小括號) 的輸入都要「成對」, 尤其在一長串公式輸入時, 未免亂掉, 所以採成對方式輸入, 再將游標移入括號中, 補其中內容, 待完整後再將游標移出括號外。

步驟 3: 中間 5 個儲存格當然有 5 個元素, 每個元素以「逗號」隔開, 其中不需要空格, 所以就為 “1, 2, 3, 4, 5”。

最後步驟 4: 就是陣列輸入與運算的必要動作 $\widehat{\text{CSE}}$ 。

圖的第 3 個步驟的資料編輯列,

$$= \{1, 2, 3, 4, 5\}$$

是我們所輸入的符號與數字, 但到了第 4 個步驟, 鍵入 $\widehat{\text{CSE}}$ 後, 資料編輯列變成

$$\{= \{1, 2, 3, 4, 5\}\}$$

比較 3、4 步驟的差異, 就是在我們輸入的一串符號與數字外面多了一組大括號。這個大括號是 $\widehat{\text{CSE}}$ 所產生的, 並不是我們輸入的, 也就是象徵陣列的輸入與運算最重要的程序。另外回去任選一個儲存格, 你都會在資料編輯列看到 $\{= \{1, 2, 3, 4, 5\}\}$, 每個儲存格都一樣。


這裡另外要提出幾點注意事項:

1. 選取儲存格要連續, 如這裡的例子是 A1 到 E1, 5 個儲存格連續; 不能僅選取 {A1, C1, D1, E1}, 跳開 B1 不選。
2. 我們要在鍵入 = 後, 接一個大括號, 如果這個大括號沒有鍵入, 會產生錯誤。
3. 逗號是分隔此列不同「欄」的標示, 至於同欄不同「列」的分隔就是下節要介紹的。

若是 A1 到 E1 格中有一格 B1 儲存格是空的, 我們可以把 B1 那個元素以二個雙引號代替, 其代表空值, 就是輸入

= {1, "", 3, 4, 5}

如圖,




	A	B	C	D	E	F
1	1		3	4	5	
2						
3						
4						

Excel 會自動跳過第 2 個元素不輸入。但要注意的是 B1 儲存格仍是 A1 到 E1 同一體系中的一員, 無法單獨另外被改變。

如果你輸入

= {1, , 3, 4, 5}


二個逗號間沒有留空白, Excel 並不承認它是空值, 所以會產生錯誤。



但是若是你輸入

= {1, ,3,4,5}

二個逗號間留一個空白鍵, 雖然 Excel 是會接受, 也會產生一個結果;



The screenshot shows the Excel interface. The formula bar contains the formula `{=1," ,3,4,5}`. The spreadsheet grid shows the following data:

	A	B	C	D	E	F
1	1		3	4	5	
2						


但這個結果的 B1 儲存格就不再是個空值, 而是一個空白鍵符號, 空值與空白鍵符號是有差別的, 尤其是在運算的時候, 空值是可以運算的, 空白鍵卻不可以加以運算。

4.2.2 陣列輸入一欄

列輸入時用「逗號」隔開, 如前節所介紹, 而欄 (行) 輸入時就要用「分號」隔開。

1. 選取 A2 到 A6 儲存格
2. 鍵入 = {6;7;8;9;10}
3. 然後按 **Ctrl** + **Shift** + **Enter**

你會得到 A2 到 A6 為 6 ~ 10 的一欄 (行) 陣列。



The screenshot shows the Excel interface. The formula bar contains the formula `{=6;7;8;9;10}`. The spreadsheet grid shows the following data:

	A	B	C	D	E	F
1						
2	6					
3	7					
4	8					
5	9					
6	10					
7						

一樣, 每格儲存格的內容都是 `{=6;7;8;9;10}`。

同上, 等號外的大括號是 **CSE** 產生的, 等號內的大括號是我們鍵進去的, 而且不可或缺。欄的列分隔是分號, 所以如果你是用逗號, 就錯了。

讓我們來看看若是把分號鍵成逗號, 會發生什麼事。

取得外部資料			連線			
B2		fx {={1,2,3,4,5}}				
	A	B	C	D	E	F
1						
2	6	1				
3	7	1				
4	8	1				
5	9	1				
6	10	1				

當我們選取一欄，卻鍵入以逗號分隔各元素時，如上圖所看到，每格都僅會出現第 1 個元素“1”。同理，當選取一行，卻鍵入以分號分隔其中元素時，依舊只會產生第 1 個元素在各儲存格。

我們有了逗號分隔欄，又有了分號分隔列，這二個湊起來就可以進行一個大於 1 的陣列了。

4.2.3 陣列輸入—列欄

行列輸入時，同一列用逗號分隔，同一行用分號分隔，選擇一個 4×3 的區塊，輸入

$$= \{1, 3, 5; 7, 9, 11; 13, 15, 17; 19, 21, 23\}$$

再同時按 **Ctrl** + **Shift** + **Enter**，就得到一個 3 行 4 列的陣列。

取得外部資料			連線		排序與篩		
A1		fx {={1,3,5;7,9,11;13,15,17;19,21,23}}					
	A	B	C	D	E	F	G
1	1	3	5				
2	7	9	11				
3	13	15	17				
4	19	21	23				

如果你把上式輸入改成

$$= \{1; 7; 13; 19, 3; 9; 15; 21, 5; 11; 17; 23\}$$

一欄一欄輸入，Excel 是不接受的，一定是要一列輸入完畢，用分號隔開再輸入一列。

陣列的輸入是一整體視為一個系統，當你選取某個 $n \times m$ 儲存格，則一整個 $n \times m$ 是同一個體系，不能單獨更改其中某（幾）儲存格；只能再全選一次，在資料編輯列更改，再一次 **CSE**。

陣列的輸入要鍵入左右大括弧,還要鍵 $\widehat{\text{CSE}}$,說到底並不會比單格(一維度)的輸入方便,所以很少人知道,知道後也很更少人會用,就連我也不用,只是課程要往下之前,必要把脈絡交待清楚。

第 5 章

基本陣列運算

Excel 運算通常都是一維度，也就是使用上儲存格 (視覺格子) 的定義，如 A_1+A_2 ，或者是一欄 (列) 加總另一欄 (列) 到「一個」儲存格，只要結果是「一格」時，應該都不算困難，通常在 Excel 的函數組合後，應該都可以得到答案，當然這要對 Excel 函數有點精通。但當結果是「多種 (格)」時，就變得困難，這就是陣列運算存在的價值。

陣列運算可以產生多格的結果，或運算時有碰觸到陣列運算等等，在一維沒有辦法處理的運算，通通可以由陣列運算處理，也包含樞紐分析表、資料分析等方法，全都是要解決多重結果的問題，一樣都可由陣列運算包辦。

另外，Excel 對上數學線性代數的矩陣運算也有困難，特別是在乘法，以下會看到數學上的矩陣與 Excel 的陣列有不同的定義。

5.1 數學上矩陣運算

Excel 陣列運算與數學上的矩陣計算還是不一樣，所以進入陣列運算之前，我們先回顧一下數學上的矩陣運算。

5.1.1 矩陣運算-加法

矩陣加法的限制是兩個矩陣的維度要一樣，

- $\forall \mathbf{A}_{i \times j} + \mathbf{B}_{m \times n} = \mathbf{C}_{p \times q}, i = m = p, j = n = q$
- $\mathbf{A} + \mathbf{0} = \mathbf{A}_{p \times q}$
- $\mathbf{A} + \mathbf{B} = \mathbf{B} + \mathbf{A}$
- $(\mathbf{A} + \mathbf{B}) + \mathbf{C} = \mathbf{A} + (\mathbf{B} + \mathbf{C})$
- $(\mathbf{A} + \mathbf{B})' = \mathbf{A}' + \mathbf{B}'$

例 1: $\mathbf{A}_{(2 \times 2)} + \mathbf{B}_{(2 \times 2)}$ 得到 $\mathbf{C}_{(2 \times 2)}$:

$$\begin{bmatrix} \alpha & \beta \\ \gamma & \delta \end{bmatrix}_{2 \times 2} + \begin{bmatrix} a & b \\ c & d \end{bmatrix}_{2 \times 2} = \begin{bmatrix} \alpha + a & \beta + b \\ \gamma + c & \delta + d \end{bmatrix}_{2 \times 2}$$

看看實例,

$$\begin{bmatrix} 1 & 3 \\ 2 & 4 \end{bmatrix} + \begin{bmatrix} 5 & 7 \\ 6 & 8 \end{bmatrix} = \begin{bmatrix} 6 & 10 \\ 8 & 12 \end{bmatrix}$$

例 2: $\mathbf{A}_{(3 \times 2)} + \mathbf{B}_{(3 \times 2)}$ 得到 $\mathbf{C}_{(3 \times 2)}$:

$$\begin{bmatrix} \alpha & \beta \\ \gamma & \delta \\ \epsilon & \zeta \end{bmatrix}_{3 \times 2} + \begin{bmatrix} a & b \\ c & d \\ e & f \end{bmatrix}_{3 \times 2} = \begin{bmatrix} \alpha + a & \beta + b \\ \gamma + c & \delta + d \\ \epsilon + e & \zeta + f \end{bmatrix}_{3 \times 2}$$

看看實例,

$$\begin{bmatrix} 1 & 4 \\ 2 & 5 \\ 3 & 6 \end{bmatrix} + \begin{bmatrix} 7 & 10 \\ 8 & 11 \\ 9 & 12 \end{bmatrix} = \begin{bmatrix} 8 & 14 \\ 10 & 16 \\ 12 & 18 \end{bmatrix}$$

5.1.2 矩陣運算-乘法

- $\forall \mathbf{A}_{i \times j} \mathbf{B}_{m \times n} = \mathbf{C}_{p \times q}, i = n = p = q, j = m$
- $\mathbf{A} \mathbf{o} = \mathbf{o}$
- $\mathbf{A} \mathbf{I} = \mathbf{A}$
- $\mathbf{A} \mathbf{B} \neq \mathbf{B} \mathbf{A}$
- $(\mathbf{A} \mathbf{B}) \mathbf{C} = \mathbf{A} (\mathbf{B} \mathbf{C})$
- $\mathbf{A} (\mathbf{B} + \mathbf{C}) = \mathbf{A} \mathbf{B} + \mathbf{A} \mathbf{C}$
- $(\mathbf{A} \mathbf{B})' = \mathbf{B}' \mathbf{A}'$

例 3: $\mathbf{A}_{(2 \times 2)} \mathbf{B}_{(2 \times 2)}$ 得到 $\mathbf{C}_{(2 \times 2)}$:

$$\begin{bmatrix} \alpha & \beta \\ \gamma & \delta \end{bmatrix}_{2 \times 2} \begin{bmatrix} a & b \\ c & d \end{bmatrix}_{2 \times 2} = \begin{bmatrix} \alpha a + \beta c & \alpha b + \beta d \\ \gamma a + \delta c & \gamma b + \delta d \end{bmatrix}_{2 \times 2}$$

看看實例,

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix} \begin{bmatrix} 5 & 6 \\ 7 & 8 \end{bmatrix} = \begin{bmatrix} 5+14 & 6+16 \\ 5+28 & 18+32 \end{bmatrix} = \begin{bmatrix} 19 & 22 \\ 43 & 50 \end{bmatrix}$$

例 4: $\mathbf{A}_{(2 \times 3)} \mathbf{B}_{(3 \times 2)}$ 得到 $\mathbf{C}_{(2 \times 2)}$:

$$\begin{bmatrix} \alpha & \beta & \gamma \\ \delta & \epsilon & \zeta \end{bmatrix}_{2 \times 3} \begin{bmatrix} a & b \\ c & d \\ e & f \end{bmatrix}_{3 \times 2} = \begin{bmatrix} \alpha a + \beta c + \gamma e & \alpha b + \beta d + \gamma f \\ \delta a + \epsilon c + \zeta e & \delta b + \epsilon d + \zeta f \end{bmatrix}_{2 \times 2}$$

看看實例,

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix} \begin{bmatrix} 7 & 10 \\ 8 & 11 \\ 9 & 12 \end{bmatrix} = \begin{bmatrix} 7+16+27 & 10+22+36 \\ 28+40+54 & 40+55+72 \end{bmatrix} = \begin{bmatrix} 50 & 68 \\ 122 & 167 \end{bmatrix}$$

若把二個矩陣掉換次序,

$$\begin{bmatrix} 7 & 10 \\ 8 & 11 \\ 9 & 12 \end{bmatrix} \begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \end{bmatrix} = \begin{bmatrix} 7+40 & 14+50 & 21+60 \\ 8+44 & 16+55 & 24+66 \\ 9+48 & 18+60 & 27+72 \end{bmatrix} = \begin{bmatrix} 74 & 64 & 81 \\ 52 & 71 & 90 \\ 57 & 78 & 99 \end{bmatrix}$$

只要順序掉換, 整個乘積的結果是完全不一樣的。

5.2 Excel 陣列運算

前面介紹了數學上的矩陣運算, 原則上數學的矩陣運算只有加、減 (負數的加法)、乘法, 並沒有除法, 詳細見數學課本。

但是這裡要介紹的 Excel 陣列運算其實是相對應儲存格的加、減、乘、除, 只要除數不為 0, 通通都可以計算, 接著我們來介紹一些陣列的簡單四則運算。

5.2.1 Excel 陣列運算—加法

Excel 的陣列加法運算和矩陣運算一樣,

$$\begin{bmatrix} \alpha & \beta \\ \gamma & \delta \end{bmatrix}_{2 \times 2} + \begin{bmatrix} a & b \\ c & d \end{bmatrix}_{2 \times 2} = \begin{bmatrix} \alpha + a & \beta + b \\ \gamma + c & \delta + d \end{bmatrix}_{2 \times 2}$$

$A_{(2 \times 2)} + B_{(2 \times 2)}$ 會得到 $C_{(2 \times 2)}$, 若 $A_{(3 \times 2)} + B_{(3 \times 2)}$ 會得到 $C_{(3 \times 2)}$ 。

我們來舉實例,

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}_{2 \times 2} + \begin{bmatrix} 6 & 8 \\ 10 & 12 \end{bmatrix}_{2 \times 2} = \begin{bmatrix} 7 & 10 \\ 13 & 16 \end{bmatrix}_{2 \times 2}$$

傳統的一維運算方式,

1. 在 A1 到 B2 各別輸入 1, 2, 3, 4,
2. 再 D1 到 E2 各別輸入 6, 8, 10, 12,
3. 然後在 G1 格輸入 “=A1+D1”,
4. 按 得到 G1 格的結果後,
5. 複製到 G2、H1、H2。

如圖,

	A	B	C	D	E	F	G	H	I
1	1	2		6	8		7	10	
2	3	4		10	12		13	16	

以上方法應該是大家熟悉的, 所以不再多所著墨。我們重點在陣列的方式, 先在 A1 到 B2 輸入 $=\{1, 2; 3, 4\}$, 再鍵入 \widehat{CSE} , 得到第 1 個陣列, 當然你要每個單格輸入, 鍵入 也可以; 依此類推, 在 D1 到 E2 輸入 $=\{6, 8; 10, 12\}$, 再 \widehat{CSE} , 得到第 2 個陣列。我們要算的是二個陣列相加, 假設要將結果顯示在 G1 到 H2, 步驟如下,

1. 選取儲存格 G1 到 H2,
2. 鍵入 “=”,
3. 選取第 1 個陣列 A1:B2,
4. 鍵入 “+”,
5. 選取第 2 個陣列 D1:E2,
6. 鍵入 \widehat{CSE} 。

即為所得。其中第 2~5 步驟也可以是用鍵入

$$= A1:B2+D1:E2$$

的方式，一樣會得到相同的結果。

圖解如下，

取得外部資料			連線			排序與篩選			
G1			fx {=A1:B2+D1:E2}						
	A	B	C	D	E	F	G	H	I
1	1	2		6	8		7	10	
2	3	4		10	12		13	16	

看圖中的資料編輯列，

$$\{= A1:B2+D1:E2\}$$

最外面有一個大括弧，這個大括弧是 $\widehat{\text{CSE}}$ 產生的，並不是我們輸入的，代表陣列運算的意思。

5.2.2 Excel 陣列運算—乘法

但是矩陣運算與陣列乘法就不同，一個矩陣 $n \times m$ 要能乘上另一個矩陣，其維度必須要 $m \times n$ ，乘之後的積矩陣維度必為 $n \times n$ 。

Excel 陣列乘法與矩陣乘法不同，陣列乘法的二個陣列維度要相同，乘完之後的積為維度也與二乘數的維度相同，如

$$\begin{bmatrix} \alpha & \beta \\ \gamma & \delta \\ \epsilon & \zeta \end{bmatrix}_{3 \times 2} * \begin{bmatrix} a & b \\ c & d \\ e & f \end{bmatrix}_{3 \times 2} = \begin{bmatrix} \alpha a & \beta b \\ \gamma c & \delta d \\ \epsilon e & \zeta f \end{bmatrix}_{3 \times 2}$$

一樣來舉實例看看，

$$\begin{bmatrix} 1 & 4 \\ 2 & 5 \\ 3 & 6 \end{bmatrix}_{3 \times 2} * \begin{bmatrix} 7 & 10 \\ 8 & 11 \\ 9 & 12 \end{bmatrix}_{3 \times 2} = \begin{bmatrix} 7 & 40 \\ 16 & 55 \\ 27 & 72 \end{bmatrix}_{3 \times 2}$$

一維的方式相信不用再介紹，直接介紹陣列的方式。先在 A1 到 B3 輸入 $=\{1, 4; 2, 5; 3, 6\}$ ，再鍵入 $\widehat{\text{CSE}}$ ，得到第 1 個陣列，當然你要每個單格輸入，鍵

入 **Enter** 也可以; 依此類推, 在 D1 到 E3 輸入 = {7, 10; 8, 11; 9, 12}, 再 **CSE**, 得到第 2 個陣列。我們要算的是二個陣列相乘, 假設要將結果顯示在 G1 到 H3, 步驟如下,

1. 選取儲存格 G1 到 H3,
2. 鍵入 “=”,
3. 選取第 1 個陣列 A1:B3,
4. 鍵入 “*”,
5. 選取第 2 個陣列 D1:E3,
6. 鍵入 **CSE**。

即為所得。其中第 2~5 步驟也可以是用鍵入

$$= A1:B3 * D1:E3$$

的方式, 一樣會得到相同的結果。

圖解如下,

取得外部資料			連線			排序與篩選			
G1			fx (=A1:B3*D1:E3)						
	A	B	C	D	E	F	G	H	I
1	1	4		7	10		7	40	
2	2	5		8	11		16	55	
3	3	6		9	12		27	72	
4									

Excel陣列運算是相對應儲存格的運算, 意思是 G1 格的乘積其實是 $A1 * D1$, 也就是二個陣列的第一個元素運算。當我們了解至此, 可以很容易開始延伸四則運算, 而且一次可以運算好幾個陣列, 只要彼此陣列的維度相同, 就可以輕鬆的運算。

5.3 基本綜合陣列運算

5.3.1 多重結果

多陣列運算例子, 如

$$\begin{bmatrix} 1 & 4 \\ 2 & 5 \\ 3 & 6 \end{bmatrix} + \begin{bmatrix} 7 & 10 \\ 8 & 11 \\ 9 & 12 \end{bmatrix} * \begin{bmatrix} 13 & 16 \\ 14 & 17 \\ 15 & 18 \end{bmatrix} - \begin{bmatrix} 19 & 22 \\ 20 & 23 \\ 21 & 24 \end{bmatrix} = \begin{bmatrix} 73 & 142 \\ 94 & 169 \\ 117 & 198 \end{bmatrix}$$

圖解如下，

取得外部資料				連線				排序與篩選							
M1				fx				{=A1:B3+D1:E3*G1:H3-J1:K3}							
	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1	1	4		7	10		13	16		19	22		73	142	
2	2	5		8	11		14	17		20	23		94	169	
3	3	6		9	12		15	18		21	24		117	198	
4															

運算的儲存格 M1 到 N3 都是顯示為

$$\{=A1:B3+D1:E3*G1:H3-J1:K3\}$$

其中 M1 儲存格 =A1+D1*G1-J1，所以 $1+7*13-19=73$ 。M2、M3、N1、N2、N3 等皆是相對應格的運算，請注意，四則運算是先乘除後加減。

結合上一章的陣列輸入原則，要手動鍵入一個大括號 {}，我們把運算與輸入融合一起操作。假設 A1 到 E1 已分別為 1, 2, 3, 4, 5 了，

剪貼簿		字型				
E1		fx				
	A	B	C	D	E	F
1	1	2	3	4	5	
2						

我們要用陣列運算方式在 A2 到 E2 分別計算成 $1^2, 2^2, 3^2, 4^2, 5^2$ 。

方法 1: 在 A2 到 E2 輸入 “=A1:E1*A1:E1”，鍵入 $\overbrace{\text{CSE}}$ 。

剪貼簿		字型				
E1		fx				
	A	B	C	D	E	F
1	1	2	3	4	5	
2						

方法 2: 在 A2 到 E2 輸入 “=A1:E1^2”，鍵入 $\overbrace{\text{CSE}}$ 。

剪貼簿		字型				
A2		fx				
	A	B	C	D	E	F
1	1	2	3	4	5	
2	1	4	9	16	25	

方法 3: 在 A2 到 E2 輸入 “=A1:E1*{1, 2, 3, 4, 5}”，鍵入 $\overbrace{\text{CSE}}$ 。

	A	B	C	D	E	F
1	1	2	3	4	5	
2	1	4	9	16	25	
3						

方法 3 就是結合輸入方式, 手動創造一個大括號內的陣列去運算。

上面是列輸入兼列運算, 若是欄輸入兼欄運算, 則要用分號, 試著在 B1 到 B5 計算 $6^2, 7^2, 8^2, 9^2, 10^2$,

$$= \{6; 7; 8; 9; 10\} * A1 : A5$$

圖解

	A	B	C	D	E	F
1	6	36				
2	7	49				
3	8	64				
4	9	81				
5	10	100				

特別注意, 當你使用欄運算, 卻輸入逗號分開如 $= \{6, 7, 8, 9, 10\} * A1:A5$, 會得到 (36, 42, 48, 54, 60) 的陣列,

	A	B	C	D	E	F
1	6	36				
2	7	42				
3	8	48				
4	9	54				
5	10	60				

它不是我們要正確解答, 是 A1 到 A5 分別乘以第一個元素 6 的結果。會得到這樣的結果, 是因為 Excel 判斷 1 欄只能乘以 1 欄, 所以只選擇第一欄的 6。如果想要得到正確答案, 要加 TRANSPOSE() 函數修正,

$$= \text{TRANSPOSE}(\{6, 7, 8, 9, 10\}) * A1:A5$$

經過 TRANSPOSE() 函數轉置後, 才會是我們要的答案。

		B1		fx {=TRANSPOSE((6,7,8,9,10)*A1:A5)}				
	A	B	C	D	E	F	G	
1	6	36						
2	7	49						
3	8	64						
4	9	81						
5	10	100						

最後，當然就是陣列輸入兼陣列運算了。在 A1 到 C3 格中運算 1 ~ 9 的平方，

		A1			fx {=(1,2,3;4,5,6;7,8,9)^2}		
	A	B	C	D	E	F	
1	1	4	9				
2	16	25	36				
3	49	64	81				

運用剛才學的轉置函數 = TRANSPOSE(.),

$$= \text{TRANSPOSE}(\{1, 2, 3; 4, 5, 6; 7, 8, 9\})^2$$

看看是不是會欄列轉置，

		A1			fx {=TRANSPOSE({1,2,3;4,5,6;7,8,9})^2}				
	A	B	C	D	E	F	G	H	
1	1	16	49						
2	4	25	64						
3	9	36	81						

5.3.2 單一結果

陣列運算強大的地方除了前面介紹的產出多重結果，就是過程中處理陣列，而最後產出一個單格結果，雖然是單格結果，確是一維 Excel 所不能觸及的。

介紹至此，我們只能很簡易展示其功能，真正的強大可能要在往後幾章才有辦法真正顯現。

假設我們要計算今天的營業收入，我們有 A 欄商品單價，B 欄為今天售出數量，所以今天的收入就是 A 欄 × B 欄後，加總起來。實例如圖，

	A	B	C	D	E
1	單價	數量	單價*數量		
2	56	8	448		
3	58	2	116		
4	73	1	73		
5	4	7	28		
6	7	2	14		
7	86	7	602		
8	62	0	0		
9	1	6	6		
10	34	0	0		
11	97	4	388		
12			1675		

加總後得到今天的營業收入共為到 1675 元。

若今天我們想要直接在 C12 儲存格一格算出, 不要有 C2 到 C11 格的 (單價 * 數量) 過程, 請問你有辦法嗎?

辦法就是用陣列運算,

$$= \text{SUM}(A2:A11 * B2:B11)$$

再鍵入最重要的 $\widehat{\text{CSE}}$, 結果如圖,

	A	B	C	D	E	F
1	單價	數量				
2	56	8				
3	58	2				
4	73	1				
5	4	7				
6	7	2				
7	86	7				
8	62	0				
9	1	6				
10	34	0				
11	97	4				
12			1675			
13						

一樣得到答案 1675 元。

在這裡只是牛刀小試, 其實陣列的輸入、運算很不人性, 要輸入大括號, 又要用逗號、分號來區隔欄、列, 之後還要同時按 $\widehat{\text{CSE}}$, 輸入錯誤時還很難更

正, 所以一直沒得到使用者的青睞, 連我也覺得不如不學。但實務的的運用就不同了, 尤其是結合邏輯函數 IF() 後, 在往後的幾章慢慢能體驗出來, 到時候就可以很輕易地千變萬化, 打敗樞紐分析表綽綽有餘。

第 6 章

陣列函數運算

=VLOOKUP(A2,{"電腦軟體應用", 800; "電腦硬體裝修", 1200; "工業電子", 1100; "數位電子", 1300}, 2, FALSE)

Exercise 6.1:

整數 1 加至 100 的總和 (= ROW() 用法)

Solution 6.1:

{= SUM(ROW(1 : 100))}

Exercise 6.2:

整數 1 加至 100 的總和 (= ROW() 用法)

Solution 6.2:

{= SUM(ROW(1 : 100))}

第 7 章

{=SUM(IF())}

如同前 ?? 介紹，我們虛擬創造出 99 筆資料，個筆資料有 5 個變數，身分證號 (id)、年齡 (age)、婚姻 (married)、消費日 (consumer_date)、消費金額 (amount) 等。

現在假設我們已有 99 筆消費資料，欄位有身分證號 (id)、年齡 (age)、婚姻 (married)、消費日 (consumer_date)、消費金額 (amount) 等 5 個變數。


id	age	married	consumer_date	amount
A169001035	63.67	1	2017/7/17	189437
B249356283	43.93	1	2017/6/2	328347
C151375012	49.00	1	2017/3/22	67319
D271532802	24.73	1	2016/4/15	99286
⋮	⋮	⋮	⋮	⋮
J291220778	58.17	1	2016/11/13	32

另外，爲了後面要計算「一年內」，所以不論今日是何夕，我們假設今天爲 2017/11/5，往後的計算就以它爲基準，來計算一年內。

7.1 資料設定

資料內容就如上所述，但我們來做一點手腳讓接下來的操作可以方便一點——命名，我們把 [\$A\$2:\$A\$100] 命名爲“id”，[\$B\$2:\$B\$100]、[\$C\$2:\$C\$100]、[\$D\$2:\$D\$100]、[\$E\$2:\$E\$100] 分別命名爲“age, married, date, amount”，讓往後的操作“id”就可以取代 [\$A\$2:\$A\$100]，請看以下設定。

選取 [A1:E100] 整個資料區塊，按 Ctrl + Shift + F3 鍵，跳出一個「以選取範圍建立名稱」對話框，預設爲前二項勾選，我們以表頭第一列爲建立資料名稱，所以保留「頂端列(T)」選項，取消「最左欄(L)」勾選，如圖




按確定後完成命名。


經過 Ctrl + Shift + F3 的程序，我們已經定義好名稱了，我們來看一下所命名的成果，點選 Excel 工具區的

公式 | 名稱管理員


應該可以看到名稱管理員已經定義好變數，



依名稱管理員視窗所示, 名稱爲: age, amount, consumer_date, id, married 等 5 個名稱, 而其中 consumer_date 與我們設定的 date 不同, 所以滑鼠選 consumer_date 再按編輯,



將名稱改成 date, 再確定後, 我們的命名動作就完成了。



經此設定, age, amount, date, id, married 等 5 個變數都是絕對位址的定義; age 就代表 [\$B\$2:\$B\$100], 同理, amount 就代表 [\$E\$2:\$E\$100], date 就代表 [\$D\$2:\$D\$100], id 就代表 [\$A\$2:\$A\$100], married 就代表 [\$C\$2:\$C\$100],

輸入時清楚明瞭,且不會因為滑鼠拖曳跑掉,更重點是呼叫方便,畢竟輸入 id 二字母比用滑鼠去選 [B2:B100] 方便許多。

7.2 問題設定

以上資料與設定完成後,我們要進入使用它來運算,我們先以以下 9 道問題循序漸進來學習陣列運算,

1. 多少男性消費
2. 男性消費金額
3. 已婚男性消費金額
4. 已婚男性一年內消費金額
5. 49 歲已婚男性一年內消費金額
6. 20~30 歲已婚男性一年內消費金額
7. 20~30 及 40~50 歲男性一年內消費金額
8. 20~30 歲女性及 40~50 歲男性一年內消費金額
9. 20~30 歲女性及 40~50 歲未婚男性一年內消費金額

G	H	I
		今天是 2017/11/5
解答		題目
	#1	多少男性消費
	#2	男性消費金額
	#3	已婚男性消費金額
	#4	已婚男性一年內消費金額
	#5	49歲已婚男性一年內消費金額
	#6	20~30歲已婚男性一年內消費金額
	#7	20~30及40~50歲男性一年內消費金額
	#8	20~30歲女性及40~50歲男性一年內消費金額
	#9	20~30歲女性及40~50歲未婚男性一年內消費金額

以上這 9 道題目每個問題都是 1 個儲存格能解出答案的,所以以下演示都在 G 欄中的單格解答,對於原始資料不必另外加欄位輔助運算,這就是陣列運算的強大。

若是用樞紐分析表可能有幾題要做 2、3 次樞紐轉 2、3 個彎才有辦法得出答案,最後還得自己手動加總,所以我總認為樞紐實在不好用。

7.3 前提

第 1 題, 有多少男性消費, 利用 id 的第 2 碼判定性別, 男性是 1, 女性為 2。

如 ??, 用 =MID() 函數取 id 變數的第 2 碼出來判斷,

=MID()

fx

從文字串的某個起始位置到指定長度之間的所有字元。


= MID(text, start_num, num_chars)

其中,

- text: 我們所要抽選之字元的文字串。
- start_num: 我們所要抽取的第一個字元的位置。
- num_chars: 指定我們要用此函數來抽選的字元個數。

假設我們要判斷上述資料表第一筆 id 是男是女, 用 =MID() 函數,

= MID(A2, 2, 1)



	A	B	C	D	E	F
1	id	age	married	consumer_date	amount	
2	A169001035	63.67	1	2017/7/17	189437	1


[A2] 格的字串為 A169001035, 取其中第 2 字元起的 1 個字元, 自然會回傳“1”。與

= MID(“A169001035”, 2, 1)

相同, 都會得到「字串“1”」。但此“1”非彼 1, 此“1”是文字的“1”, 非數字的 1。字串與數字的差別在於可不可以運算, 字串不能運算, 數字才可以; 若要判斷是字串或文字最簡單的方式就是在 Excel 儲存格沒有設定狀況下, 靠左的是字串, 靠右就是數字。通常我們把數字的字串用上雙引號括起來“”表示, 若是數字則保留原來的數字格式, 而一般文字就只是文字, 不能拿來計算。

其實要把“1”變成 1 很容易, 只要把“1”*1 就會變成 1, 如

= MID(A4, 2, 1) * 1



	A	B	C	D	E	F
1	id	age	married	consumer_date	amount	
2	A169001035	63.67	1	2017/7/17	189437	1
3	B249356283	43.93	1	2017/6/2	328347	2
4	C151375012	49.00	1	2017/3/22	67319	1
5	D271532802	24.73	1	2016/4/15	99286	

得到一個靠右的 1, 特別注意, 此 1 非彼“1”, 這個 1 是可以計算的數字。分辨文字、數字最主要原因是接下來的輸入與運算, 如同前面所說得 TRUE 可以當 1, FALSE 可以當 0 一樣。

用 =MID() 函數判斷男女後, 可以用 =COUNTIF() 函數 (見 ??) 計算個數,

=COUNTIF()

fx

計算範圍內符合指定條件儲存格的數目。

= COUNTIF(range, criteria)

其中,

- range: 欲計算符合給定條件儲存格數目的範圍。
- criteria: 為比較的條件, 條件可以是數字、表示式或文字, 用以指定哪些儲存格會被計算。

這邊要留意, 當 criteria 為字串時, 別忘記將文字加上雙上引號 ""。

我們把 =MID() 函數檢驗放在 [F2:F100], 先在 F2 輸入 = MID(A2, 2, 1), 複製到 F3 至 F100 儲存格, 以得到 id 的第二碼。

	A	B	C	D	E	F
1	id	age	married	consumer_date	amount	
2	A169001035	63.67	1	2017/7/17	189437	1
3	B249356283	43.93	1	2017/6/2	328347	2
4	C151375012	49.00	1	2017/3/22	67319	1
5	D271532802	24.73	1	2016/4/15	99286	2
6	E154140527	2.27	0	2016/12/3	20348	1
7	F200381688	23.32	0	2016/12/26	42331	2
8	G110652243	39.28	0	2017/10/25	15892	1
9	H237507560	70.42	1	2017/7/11	71605	2

再用 =COUNTIF() 來計算 [F2:F100] 中屬“1”的個數, 在欲求解的儲存格如下圖 L2 格輸入:

= COUNTIF(F2:F100, “1”)

	A	B	C	D	E	F	J
1	id	age	married	consumer_date	amount		
2	A169001035	63.67	1	2017/7/17	189437	1	51
3	B249356283	43.93	1	2017/6/2	328347	2	

得到 51 位男性, 即為所求。用 J3 儲存格來求女性數量:

= COUNTIF(F2:F100, “2”)

	A	B	C	D	E	F	J
1	id	age	married	consumer_date	amount		
2	A169001035	63.67	1	2017/7/17	189437	1	51
3	B249356283	43.93	1	2017/6/2	328347	2	48

得到 48 位女性。51 為男性加上 48 位女性共為 99 位, 即為我們 99 筆資料。

像這麼簡單的題目可以單用一個函數解決, 但這還不是我們這章的重點, 只是先用一般函數來做對比, 接著我們就要進入「陣列」運算, 從這個簡單題目入手, 讓大家比較容易可以理解陣列的運作。

這個例子中, 要用到 F 欄來輔助先取 id 的第 2 碼, 先區分男女, 再計算男性有多少, 進入我們的陣列運算就不用。

7.4 $\overbrace{\hspace{1cm}}$ CSE

$\overbrace{\hspace{1cm}}$ 的意思是 $\boxed{\text{Ctrl}}$ + $\boxed{\text{Shift} \uparrow}$ + $\boxed{\text{Enter}}$ ，這是陣列運算 enter 指令，另一方面，單就輸入來說就是在算式的最外面產生大括號 {}。大括號在等號的外面便是 $\overbrace{\hspace{1cm}}$ 產生的，若是大括號在等號後面 (裡面)，則是爲了輸入陣列元素而鍵進去的。

以下開始真正進入陣列運算。

7.4.1 多少男性消費

$\overbrace{\hspace{1cm}}$ CSE
[A]

多少男性消費?

#1 解答:

{= SUM(IF(MID(id, 2, 1)="1", 1, 0))}

G4		fx {=SUM(IF(MID(id,2,1)="1",1,0))}						
	A	B	C	D	E	F	G	H
1	id	age	married	consumer_date	amount			
2	A169001035	63.67	1	2017/7/17	189437			今天是 2017/11/5
3	B249356283	43.93	1	2017/6/2	328347		解答	
4	C151375012	49.00	1	2017/3/22	67319		51	#1 多少男性消費
5	D271532802	24.73	1	2016/4/15	99286			#2 男性消費金額
6	E154140527	2.27	0	2016/12/3	20348			#3 已婚男性消費金額

要注意 =MID() 得出來的是文字，所以要 = 後面要加雙上引號 "1"。接著來拆解上式，

$$\begin{array}{c} \left[\begin{array}{c} A169001035 \\ B249356283 \\ C151375012 \\ D271532802 \\ \vdots \\ H111143673 \\ J291220778 \end{array} \right] \\ \text{id} \end{array} \Rightarrow \begin{array}{c} \left[\begin{array}{c} 1 \\ 2 \\ 1 \\ 2 \\ \vdots \\ 1 \\ 2 \end{array} \right] \\ \text{MID(id, 2, 1)} \end{array} \Rightarrow \begin{array}{c} \left[\begin{array}{c} 1 \\ 0 \\ 1 \\ 0 \\ \vdots \\ 1 \\ 0 \end{array} \right] \\ \text{IF(MID())} \end{array} \Rightarrow \text{SUM} \left(\begin{array}{c} \left[\begin{array}{c} 1 \\ 0 \\ 1 \\ 0 \\ \vdots \\ 1 \\ 0 \end{array} \right] \right) \\ \text{SUM(IF(MID()))} \end{array}$$

id 用 =MID() 取第 2 碼, 得到性別, 接著用 =IF() 轉成我們要的男性, 使之為 1, 讓我們不要的女性為 0, 最後把 1 與 0 加總, 得到我們要的男性個數, 即為答案。

上面的答案可以省略 =IF() 第 3 個參數, 變成:

$$\{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = "1", 1))\}$$

G4		fx {=SUM(IF(MID(d,2,1)="1",1))}							
	A	B	C	D	E	F	G	H	
1	id	age	married	consumer_date	amount				
2	A169001035	63.67	1	2017/7/17	189437				今天是 2017/11/5
3	B249356283	43.93	1	2017/6/2	328347				解答
4	C151375012	49.00	1	2017/3/22	67319		51	#1	多少男性消費
5	D271532802	24.73	1	2016/4/15	99286			#2	男性消費金額
6	E154140527	2.27	0	2016/12/3	20348			#3	已婚男性消費金額

另外利用 “1” 轉變為 1 的技巧, 也可以寫成:

$$\{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) * 1 = 1, 1))\}$$


G4		fx {=SUM(IF(MID(d,2,1)*1=1,1))}							
	A	B	C	D	E	F	G	H	
1	id	age	married	consumer_date	amount				
2	A169001035	63.67	1	2017/7/17	189437				今天是 2017/11/5
3	B249356283	43.93	1	2017/6/2	328347				解答
4	C151375012	49.00	1	2017/3/22	67319		51	#1	多少男性消費
5	D271532802	24.73	1	2016/4/15	99286			#2	男性消費金額
6	E154140527	2.27	0	2016/12/3	20348			#3	已婚男性消費金額

以上這些方式都會得到一樣的答案。


回到前一節的 =COUNTIF() 函數, 它利用 F 欄來擷取性別, 可不可以也把 =MID() 函數納入 =COUNTIF() 中, 來試:

$$= \text{COUNTIF}(\text{MID}(\text{id}, 2, 1), "1")$$

按 [Enter] 鍵, 會得到錯誤訊息; 換鍵入 [Ctrl] + [Shift ↑] + [Enter] 的結果也一樣, 通通不接受。



=COUNTIF() 函數中不接受使用陣列運算,



=COUNTIF() 的 range 參數已可以將 =MID() 函數計算出各格的第 2 碼, 如上圖 {\"1\", \"2\", \"1\", \"2\", \"1\", ...}, 可是它就是沒辦法計算得到 COUNTIF 的結果, 硬是要用 =COUNTIF() 函數的話, 只能先另外用個輔助欄位處理要預備要用的資訊了。

我們找到了一個陣列 $\widehat{\text{CSE}}$ 的優點了, 也漸漸開始有了往下走的理由了。

7.4.2 男性消費金額

CSE
[A]

男性消費金額?

#2 解答:

{=SUM(IF(MID(id,2,1)="1",amount))}

	A	B	C	D	E	F	G	H
1	id	age	married	consumer_date	amount			
2	A169001035	63.67	1	2017/7/17	189437			今天是 2017/11/5
3	B249356283	43.93	1	2017/6/2	328347			
4	C151375012	49.00	1	2017/3/22	67319		解答	
5	D271532802	24.73	1	2016/4/15	99286		51	#1 多少男性消費
6	E154140527	2.27	0	2016/12/3	20348		1052548	#2 男性消費金額
								#3 已婚男性消費金額

只要把 IF() 的第 2 個參數換成消費金額, 讓 IF() 邏輯判斷為 TRUE 時回傳消費金額, 最後再加總就是答案了。看一下拆解,

$$\begin{array}{c} \left[\begin{array}{c} A169001035 \\ B249356283 \\ C151375012 \\ D271532802 \\ \vdots \\ H111143673 \\ J291220778 \end{array} \right] \\ \text{id} \end{array} \Rightarrow \begin{array}{c} \left[\begin{array}{c} 1 \\ 2 \\ 1 \\ 2 \\ \vdots \\ 1 \\ 2 \end{array} \right] \\ \text{MID(id, 2, 1)} \end{array} \Rightarrow \begin{array}{c} \left[\begin{array}{c} 189437 \\ \text{FALSE} \\ 67319 \\ \text{FALSE} \\ \vdots \\ 9868 \\ \text{FALSE} \end{array} \right] \\ \text{IF(MID())} \end{array} \Rightarrow \text{SUM} \left(\begin{array}{c} \left[\begin{array}{c} 189437 \\ \text{FALSE} \\ 67319 \\ \text{FALSE} \\ \vdots \\ 9868 \\ \text{FALSE} \end{array} \right] \\ \text{SUM(IF(MID()))} \end{array} \right)$$

同前, 最後被加總的置換為消費金額。id 用 =MID() 取第 2 碼, 得到性別, 接著用 =IF() 把男性轉成消費金額, 讓女性還是保留 FALSE, 最後把消費金額與 FALSE 加總, FALSE 不會計入任何數字, 所以最後加總的數字為男性消費金額, 即為答案。

若是用 *1 把文字轉為數字,

$$\{=SUM(IF(MID(id, 2, 1)*1 = 1, amount))\}$$

Excel 會把 FALSE 自動轉換成 0, 所以也不影響最後的加總, 這也就是在這個例子第 3 個參數能省略的原因。

7.4.3 已婚男性消費金額

較上題加個「已婚」的條件, 引用前面 ?? 在 =IF() 函數交集的概念, 把男性與已婚交集篩選出來。特別注意, 在一維的計算下, =AND() 或 =OR() 可以用 =AND() 與 =OR() 來處理交、聯集, 但在陣列的運算中卻無法使用, 所以我們開始要介紹交、聯集的 “*” 及 “+”。

CSE
[A]

已婚男性消費金額?

#3 解答:

$$\{=SUM(IF((MID(id, 2, 1)="1")*(married=1), amount))\}$$

G6		={SUM(IF((MID(id,2,1)="1")*(married=1),amount))}						
	A	B	插入函數	D	E	F	G	H
1	id	age	married	consumer_date	amount			
2	A169001035	63.67	1	2017/7/17	189437			今天是 2017/11/5
3	B249356283	43.93	1	2017/6/2	328347		解答	
4	C151375012	49.00	1	2017/3/22	67319		51	#1 多少男性消費
5	D271532802	24.73	1	2016/4/15	99286		1052548	#2 男性消費金額
6	E154140527	2.27	0	2016/12/3	20348		489175	#3 已婚男性消費金額
7	F200381688	23.32	0	2016/12/26	42331			#4 已婚男性一年內消費

其中 =IF() 內的 logical_test 第一個參數為

$$(MID(id, 2, 1)="1")*(married=1)$$

就是在處理「男性」和「已婚」,

1. MID(id, 2, 1)="1" 前面已交代過, 是為篩選出男性,
2. married=1 即是已婚的條件,

這 2 個條件交集起來就是「已婚的男性」, 而這 2 個條件在運算上要交集起來就是要用 “*”。

若是「已婚的男性」為真, =IF() 函數回傳消費金額, 再交給 =SUM() 函數加總, 就會是「已婚男性消費金額」問題的解了。

回到 ??, 在這裡用 =AND() 來交集男性和已婚看是否可行, 試著輸入

```
{=SUM(IF((AND(MID(id, 2, 1)="1", (married=1)), amount))}
```

結果回傳錯誤, 因此知道 =AND() 不可行, 所以乖乖使用 “*” 吧!

7.4.4 已婚男性一年內消費金額

本題加了「一年內」條件, 正常狀況下, 「一年內」用 =TODAY()¹ 函數減去消費日期在一年 365 天內,

```
TODAY()-date<365
```

就為「一年內」條件。

<code>=TODAY()</code>	<i>fx</i>
傳回格式為日期的目前日期。	
<hr/>	
<code>= TODAY()</code>	
其中並不需要引數、參數	

回到「已婚男性一年內消費金額」這道問題裡, 我們在本章最前面已設定今天為 2017/11/5, 所以要解問題內的「一年內」應該要

```
("2017/11/5"-date<365)
```

2017/11/5 要用引號括起來, 否則會是 2017 ÷ 11 ÷ 5, 答案會是 36.67。所以記得要有引號, 以識別為日期。

把已婚、男性、一年內, 這 3 條件用 “*” 交集起來, 同時成立時就得解了。

¹=TODAY() 函數沒有參數, 但鍵入儲存格是還是要鍵入括號 ()。此函數會回傳今天的日期當儲存格格式改成「通用」或「數值」時, 會回傳以 1900 年 1 月 1 日當第 1 天, 起算至今天的序值, 也就是以 1990/1/1 起至今經過天數; 當 =TODAT() 經過運算, 就會是序列值。

已婚男性一年內消費金額?

#4 解答:

$$\{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = "1") * (\text{married} = 1) * (\text{"2017/11/5"} - \text{date} < 365), \text{amount}))\}$$

B	C	D	E	F	G	H	I
age	married	consumer_date	amount				
63.67	1	2017/7/17	189437				今天是 2017/11/5
43.93	1	2017/6/2	328347		解答		題目
49.00	1	2017/3/22	67319		51	#1	多少男性消費
24.73	1	2016/4/15	99286		1052548	#2	男性消費金額
2.27	0	2016/12/3	20348		489175	#3	已婚男性消費金額
23.32	0	2016/12/26	42331		388466	#4	已婚男性一年內消費金額
39.28	0	2017/10/25	15892			#5	49歲已婚男性一年內消費金額

在真實日期狀況下, 本題最直接解法可以用 =TODAY() 取代 “2017/11/5”, 就會變成:

$$\{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = "1") * (\text{married} = 1) * (\text{TODAY}() - \text{date} < 365), \text{amount}))\}$$

7.4.5 49 歲已婚男性一年內消費金額

「49 歲」條件,

(age=49)

再加上原始的男性、已婚、一年內等三個條件, 交集要成立的條件來到四個,

$$(\text{MID}(\text{id}, 2, 1) = "1") * (\text{married} = 1) * (\text{"2017/11/5"} - \text{date} < 365) * (\text{age} = 49)$$

都要放入 =IF() 的第 1 個參數中, 而解題一樣是要加總消費金額。

這裡 (age=49) 的設定是年齡要剛好 49 歲整, 而一般若是 49 歲指的是: 滿 49 歲且未滿 50 歲, 不過那是下一題的範疇, 這題我們先以 (age=49) 來當條件運算。

49 歲已婚男性一年內消費金額?

#5 解答:

$$\{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = "1") * (\text{married} = 1) * (\text{"2017/11/5"} - \text{date} < 365) * (\text{age} = 49), \text{amount}))\}$$

B	C	D	E	F	G	H	I
age	married	consumer_date	amount				
63.67	1	2017/7/17	189437				今天是 2017/11/5
43.93	1	2017/6/2	328347		解答		題目
49.00	1	2017/3/22	67319		51	#1	多少男性消費
24.73	1	2016/4/15	99286		1052548	#2	男性消費金額
2.27	0	2016/12/3	20348		489175	#3	已婚男性消費金額
23.32	0	2016/12/26	42331		398334	#4	已婚男性一年內消費金額
39.28	0	2017/10/25	15892		67319	#5	49歲已婚男性一年內消費金額
79.43	1	2017/7/1	71605			#6	20~30歲已婚男性一年內消費金額

看上圖消費金額 67319 的解答，正好是第 4 列資料的金額，因他的年齡正好是 49.00。

滿 49 歲且未滿 50 歲的條件為

$$((\text{age} \geq 49) * (\text{age} < 50))$$

開始進如下題。

7.4.6 20~30 歲已婚男性一年內消費金額

如同上面所說，20~30 歲的條件是：滿(包含) 20 歲且未滿(不含) 30 歲，所以

$$((\text{age} \geq 20) * (\text{age} < 30))$$

20 歲的條件帶等號。中文敘述為 1 個條件，但算式表現出來為 2 個條件的交集，所以 =IF() 內的第 1 個參數條件已來到 5 個，

$$\text{MID}(\text{id}, 2, 1) = "1" * (\text{married} = 1) * (\text{"2017/11/5"} - \text{date} < 365) * ((\text{age} \geq 20) * (\text{age} < 30))$$

最後本題的答案,

CSE

[A]

20~30 歲已婚男性一年內消費金額?

#6 解答:

$$\{=SUM(IF((MID(id, 2, 1)="1")*(married=1)*("2017/11/5"-date<365)*((age>=20)*(age<30)), amount))\}$$

fx		{=SUM(IF((MID(id,2,1)="1")*(married=1)*("2017/11/5"-date<365)*((age>=20)*(age<30)),amount))}						
	C	D	E	F	G	H	I	
	married	consumer_date	amount					
57	1	2017/7/17	189437				今天是 2017/11/5	
93	1	2017/6/2	328347				題目	
					解答			
00	1	2017/3/22	67319		51	#1	多少男性消費	
73	1	2016/4/15	99286		1052548	#2	男性消費金額	
27	0	2016/12/3	20348		489175	#3	已婚男性消費金額	
32	0	2016/12/26	42331		388466	#4	已婚男性一年內消費金額	
28	0	2017/10/25	15892		67319	#5	49歲已婚男性一年內消費金額	
43	1	2017/7/1	71605		56974	#6	20~30歲已婚男性一年內消費金額	
91	0	2016/11/22	29241			#7	20~30及40~50歲男性一年內消費金額	

7.4.7 20~30 及 40~50 歲男性一年內消費金額

交集過後, 來了聯集。

除了 20~30 歲, 又要「聯」集 40~50 歲, 就年齡這個條件就該下

$$((age>=20)*(age<30))+((age>=40)*(age<50))$$

前半段為 20~30 歲, 後半段為 40~50 歲, 二段年齡條件用 “+” 聯集起來。

現在這道題的條件有男性 (@1)、一年內 (@1)、20~30 歲 (@2)、40~50 歲 (@2), 共 6 個條件 (暫時沒有已婚條件), 其中二段年齡的 4 個條件交聯集後, 又要與男性及一年內交集, 請注意, 這裡的年齡條件因為是 2 組各 2 條件的交集再聯集, 一定要視為一個大條件, 所以要有一個括號把聯集條件括成一個條件, 之後再與性別、期間交集, 因此 =IF() 內的第 1 個參數為

$$MID(id, 2, 1)="1")*(("2017/11/5"-date<365)*(((age>=20)*(age<30))+((age>=40)*(age<50))))$$

若沒有該括號，則會變成「20~30 歲男性一年內及 40~50 歲不分性別、消費期間」的消費金額，這樣不符這道問題的題意，所以交聯集不得不小心，下一題開始會有更多條件來練習交聯集。

CSE
[A]

20~30 及 40~50 歲男性一年內消費金額?

#7 解答:

$$\{=SUM(IF((MID(id, 2, 1)="1")*(("2017/11/5"-date<365)*(((age>=20)*(age<30))+((age>=40)*(age<50)))), amount))\}$$

fx		{=SUM(IF((MID(id,2,1)="1")*(("2017/11/5"-date<365)*(((age>=20)*(age<30))+((age>=40)*(age<50)))),amount))}				
C	D	E	F	G	H	I
married	consumer_date	amount				
1	2017/7/17	189437				今天是 2017/11/5
1	2017/6/2	328347		解答		題目
1	2017/3/22	67319		51	#1	多少男性消費
1	2016/4/15	99286		1052548	#2	男性消費金額
0	2016/12/3	20348		489175	#3	已婚男性消費金額
0	2016/12/26	42331		388466	#4	已婚男性一年內消費金額
0	2017/10/25	15892		67319	#5	49歲已婚男性一年內消費金額
1	2017/7/1	71605		56974	#6	20~30歲已婚男性一年內消費金額
0	2016/11/22	29241		161993	#7	20~30及40~50歲男性一年內消費金額
1	2017/10/10	68618			#8	20~30歲女性及40~50歲男性一年內消費金額

7.4.8 20~30 歲女性及 40~50 歲男性一年內消費金額

用性別把年齡條件拆開，20~30 歲為女性，

$$(age \geq 20) * (age < 30) * (MID(id, 2, 1) = "2")$$

40~50 歲還是原來的男性，但因為性別條件把年齡拆開，所以這組年齡條件也要安插個性別，

$$(age \geq 40) * (age < 50) * (MID(id, 2, 1) = "1")$$

以上二個條件聯集起來，

$$\begin{aligned} & ((age \geq 20) * (age < 30) * (MID(id, 2, 1) = "2")) \\ & + ((age \geq 40) * (age < 50) * (MID(id, 2, 1) = "1")) \end{aligned}$$

再與一年內交集, 才是完整條件,

$$\begin{aligned} & ("2017/11/5" - date < 365) \\ & * (((age \geq 20) * (age < 30) * (MID(id, 2, 1) = "2")) \\ & \quad + ((age \geq 40) * (age < 50) * (MID(id, 2, 1) = "1"))) \end{aligned}$$

本道題目的解法:

CSE
[A]

20~30 歲女性及 40~50 歲男性一年內消費金額?

#8 解答:

$$\{=SUM(IF(("2017/11/5" - date < 365) * (((age \geq 20) * (age < 30) * (MID(id, 2, 1) = "2")) + ((age \geq 40) * (age < 50) * (MID(id, 2, 1) = "1"))), amount))\}$$

	E	F	G	H	I
	189437				今天是 2017/11/5
	328347		解答		題目
	67319		51	#1	多少男性消費
	99286		1052548	#2	男性消費金額
	20348		489175	#3	已婚男性消費金額
	42331		388466	#4	已婚男性一年內消費金額
	15892		67319	#5	49歲已婚男性一年內消費金額
	71605		56974	#6	20~30歲已婚男性一年內消費金額
	29241		161993	#7	20~30及40~50歲男性一年內消費金額
	68618		528797	#8	20~30歲女性及40~50歲男性一年內消費金額
	65509			#9	20~30歲女性及40~50歲未婚男性一年內消費金額

7.4.9 20~30 歲女性及 40~50 歲未婚男性一年內消費金額

這道題目二邊條件不對等, 20~30 歲女性並未下婚姻條件, 只有大於 20、小於 30 歲二條件的女性;

$$(age \geq 20) * (age < 30) * (MID(id, 2, 1) = "2")$$

反觀男性,除了大於 40 歲、小於 50 歲外,還加了未婚條件,

$$(age \geq 40) * (age < 50) * (MID(id, 2, 1) = "1") * (married = 0)$$

2 個大條件聯集起來,

$$\begin{aligned} & ((age \geq 20) * (age < 30) * (MID(id, 2, 1) = "2")) \\ & + ((age \geq 40) * (age < 50) * (MID(id, 2, 1) = "1") * (married = 0)) \end{aligned}$$

外層再與「一年內」交集,

$$\begin{aligned} & ("2017/11/5" - date < 365) \\ & * (((age \geq 20) * (age < 30) * (MID(id, 2, 1) = "2")) \\ & + ((age \geq 40) * (age < 50) * (MID(id, 2, 1) = "1") * (married = 0))) \end{aligned}$$

本題最後的答案是

CSE
[A]

20~30 歲女性及 40~50 歲未婚男性一年內消費金額?

#9 解答:

$$\{=SUM(IF(("2017/11/5" - date < 365) * (((age \geq 20) * (age < 30) * (MID(id, 2, 1) = "2")) + ((age \geq 40) * (age < 50) * (MID(id, 2, 1) = "1") * (married = 0))), amount))\}$$

D	資料編輯列	E	F	G	H	I
2017/7/17	189437					今天是 2017/11/5
2017/6/2	328347			解答		題目
2017/3/22	67319			51	#1	多少男性消費
2016/4/15	99286			1052548	#2	男性消費金額
2016/12/3	20348			489175	#3	已婚男性消費金額
016/12/26	42331			388466	#4	已婚男性一年內消費金額
017/10/25	15892			67319	#5	49歲已婚男性一年內消費金額
2017/7/1	71605			56974	#6	20~30歲已婚男性一年內消費金額
016/11/22	29241			161993	#7	20~30及40~50歲男性一年內消費金額
017/10/10	68618			528797	#8	20~30歲女性及40~50歲男性一年內消費金額
2016/6/8	65509			455306	#9	20~30歲女性及40~50歲未婚男性一年內消費金額
017/10/11	67008					

7.4.10 {=SUM(IF(IF(...)))}

不習慣把「所有」條件集中在 =IF() 的第 1 個參數的人也可以把條件拆開成好幾個 =IF(IF(IF(...))), 只要條件式交集關係, 可以自由拆開或合併 =IF(), 2 個交集條件就用 2 層 =IF(IF()), 3 個交集條件就用 3 層, 依此類推。

以下我們再重新回顧一遍之前的 9 到問題, 自第 3 題起開始有 2 個條件, 所以自第 3 題開始看。

CSE	[A]
已婚男性消費金額?	
#3 解答:	
$\{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = "1") * (\text{married} = 1), \text{amount}))\}$	
另解:	
$\{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = "1", \text{IF}(\text{married} = 1, \text{amount})))\}$	

CSE	[A]
已婚男性一年內消費金額?	
#4 解答:	
$\{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = "1") * (\text{married} = 1) * (\text{"2017/11/5"} - \text{date} < 365), \text{amount}))\}$	
另解:	
$\{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = "1", \text{IF}(\text{married} = 1, \text{IF}(\text{"2017/11/5"} - \text{date} < 365, \text{amount}))))\}$	

49 歲已婚男性一年內消費金額?

#5 解答:

```
{=SUM(IF((MID(id, 2, 1)="1")*(married=1)
*("2017/11/5"-date<365)*(age=49), amount))}
```

另解:

```
{=SUM(IF(MID(id, 2, 1)="1", IF(married=1,
IF("2017/11/5"-date<365, IF(age=49, amount))))))}
```

20~30 歲已婚男性一年內消費金額?

#6 解答:

```
{=SUM(IF((MID(id, 2, 1)="1")*(married=1)
*("2017/11/5"-date<365)*((age>=20)*(age<30)), amount))}
```

另解:

```
{=SUM(IF(MID(id, 2, 1)="1", IF(married=1,
IF("2017/11/5"-date<365,
IF(age>=20, IF(age<30, amount))))))}
```

又另解:

```
{=SUM(IF(MID(id, 2, 1)="1", IF(married=1,
IF("2017/11/5"-date<365,
IF((age>=20)*(age<30), amount))))))}
```

聯集的條件可以拆成 2 個 {=SUM(IF())} 相加, 各自的 {=SUM()} 內也依自己喜好配置多個 =IF(), 來看第 7 題以後問題。

CSE

[A]

20~30 及 40~50 歲男性一年內消費金額?

#7 解答:

$$\{=SUM(IF((MID(id, 2, 1)="1")*(“2017/11/5”-date<365) *(((age>=20)*(age<30))+((age>=40)*(age<50))), amount)))\}$$

另解:

$$\{=SUM(IF(MID(id, 2, 1)="1", IF(“2017/11/5”-date<365, IF(age>=20, IF(age<30, amount)))) +SUM(IF(MID(id, 2, 1)="1", IF(“2017/11/5”-date<365, IF(age>=40, IF(age<50, amount))))))\}$$

又另解:

$$\{=SUM(IF((MID(id, 2, 1)="1")*(“2017/11/5”-date<365), IF((age>=20)*(age<30), amount))) +SUM(IF((MID(id, 2, 1)="1")*(“2017/11/5”-date<365), IF((age>=40)*(age<50), amount))))\}$$

又再另解:

$$\{=SUM(IF(MID(id, 2, 1)="1", IF(“2017/11/5”-date<365, IF(((age>=20)*(age<30))+((age>=40)*(age<50)), amount))))\}$$

20~30 歲女性及 40~50 歲男性一年內消費金額?

#8 解答:

$$\begin{aligned} & \{= \text{SUM}(\text{IF}(\text{"2017/11/5"} - \text{date} < 365) \\ & \quad * ((\text{age} \geq 20) * (\text{age} < 30) * (\text{MID}(\text{id}, 2, 1) = \text{"2"})) \\ & \quad + ((\text{age} \geq 40) * (\text{age} < 50) * (\text{MID}(\text{id}, 2, 1) = \text{"1"}))), \\ & \quad \text{amount}))\} \end{aligned}$$

另解:

$$\begin{aligned} & \{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = \text{"2"}, \text{IF}(\text{"2017/11/5"} - \text{date} < 365, \\ & \quad \text{IF}(\text{age} \geq 20, \text{IF}(\text{age} < 30, \text{amount})))) \\ & \quad + \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = \text{"1"}, \text{IF}(\text{"2017/11/5"} - \text{date} < 365, \\ & \quad \text{IF}(\text{age} \geq 40, \text{IF}(\text{age} < 50, \text{amount}))))\} \end{aligned}$$

又另解:

$$\begin{aligned} & \{= \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = \text{"2"}, \text{IF}(\text{"2017/11/5"} - \text{date} < 365, \\ & \quad \text{IF}((\text{age} \geq 20) * (\text{age} < 30), \text{amount})))) \\ & \quad + \text{SUM}(\text{IF}(\text{MID}(\text{id}, 2, 1) = \text{"1"}, \text{IF}(\text{"2017/11/5"} - \text{date} < 365, \\ & \quad \text{IF}((\text{age} \geq 40) * (\text{age} < 50), \text{amount}))))\} \end{aligned}$$

又再另解:

$$\begin{aligned} & \{= \text{SUM}(\text{IF}(\text{"2017/11/5"} - \text{date} < 365, \\ & \quad \text{IF}((\text{age} \geq 20) * (\text{age} < 30) * (\text{MID}(\text{id}, 2, 1) = \text{"2"}), \text{amount}))) \\ & \quad + \text{SUM}(\text{IF}(\text{"2017/11/5"} - \text{date} < 365, \\ & \quad \text{IF}((\text{age} \geq 40) * (\text{age} < 50) * (\text{MID}(\text{id}, 2, 1) = \text{"1"}), \text{amount}))))\} \end{aligned}$$

20~30 歲女性及 40~50 歲未婚男性一年內消費金額?

#9 解答:

$$\{=SUM(IF(("2017/11/5" - date < 365) * (((age >= 20) * (age < 30) * (MID(id, 2, 1) = "2")) + ((age >= 40) * (age < 50) * (MID(id, 2, 1) = "1")) * (married = 0)), amount))\}$$

另解:

$$\{=SUM(IF(MID(id, 2, 1) = "2", IF("2017/11/5" - date < 365, IF(age >= 20, IF(age < 30, amount)))) + SUM(IF(MID(id, 2, 1) = "1", IF("2017/11/5" - date < 365, IF(age >= 40, IF(age < 50, IF(married = 0, amount))))))\}$$

又另解:

$$\{=SUM(IF(MID(id, 2, 1) = "2", IF("2017/11/5" - date < 365, IF((age >= 20) * (age < 30), amount)))) + SUM(IF(MID(id, 2, 1) = "1", IF("2017/11/5" - date < 365, IF((age >= 40) * (age < 50) * (married = 0), amount))))\}$$

以上隨使用者最容易的邏輯思路去組合條件, 都可以得到解答。

7.5 只有 SUM 能 IF 嗎?

只有 SUM 能 IF 嗎? 當然不是。

=IF() 依條件篩選出結果後, 在其外面, 你可以任意加能運算陣列的函數。

如下:

- {=AVERAGE(IF())}

- {=MAX(IF())}
- {=MEDIAN(IF())}
- {=COUNT(IF())}
- {=LARGE(IF(),k)}

但可能要注意的是 =IF() 函數的第 3 個參數回傳值, 假如你是設定為 0, 它也是一個值, 但卻是不符合資格時 (FALSE) 的值, 若它不是你要的, 一旦外層是 =AVERAGE() 時, 它就會影響你的結果, 正確的第 3 個參數應該為 "" 或省略。

前面花很多心力在處理 =IF() 的第 1 個參數—邏輯運算式, 它是篩選、歸類 (歸為 TRUE 或 FALSE) 的過程, 但是最終還是 {=IF()} 外層的函數決定最終計算結果。{=IF()} 所產生的結果是一串數列如 {1, 0, 0, 0, 1, ...}, 外層函數就把這串數列加以運算, 拿來運算的內容就取決您讓 TRUE 與 FALSE 為何值了。

假設我銷售商品至 4 個國家, 收到各該國的貨幣貨款, A 國匯率為 2、B 國匯率為 2.8、C 國匯率為 12、D 國匯率為 8.5、銷售收入帳為:

國家 (A 欄)	貨款收入 (B 欄)
A	1848
B	1252
A	16172
C	325250
D	3723
B	669
A	5512
D	1617
D	7905
C	5360

計算以本國貨幣計之銷貨收入:

$$\{=SUM(IF(A1:A10="A", B1:B10*2, IF(A1:A10="B", B1:B10*2.8, IF(A1:A10="C", B1:B10*12, B1:B10*8.5))))\}$$

計算出答案為 4132345。這裡要講的重點是 =IF() 函數的第 2、3 個參數也都很重要, 前面所闡述的都幾乎把第 3 個參數省略, 是因為條件篩選後, 不符合條件的我們就不要了, 所以不在乎 FALSE, 直接讓它不影響我們的計算。在這裡, 我們用 3 個 =IF() 先把 A、B、C 國挑出來, 剩下的就是 D 國了, 每個 =IF() 函數都有使用滿 3 個參數。

7.6 與 =SUMIFS() 比較

以上我用 {=SUM(IF())} 陣列運算方式解了 9 道題, 每個問題的解都在一個儲存格內得到解答, 不需要額外輔助欄 (列) 前置預備運算性別、日期等。計算個 (次) 數時, 利用 =IF() 把符合資格的轉為 1, 加總後就能得到答案; 計算消費金額合計數時, 也是用 =IF() 篩選分辨條件資格, 最後加總, 而這裡的條件資格就可以是多重條件的聯集、交集等, 陣列運算的優勢便在這裡, 可以接受函數的函數運算, 因此可以很強大。不像 Excel 內建的函數 =SUMIFS(), 其參數不接受函數運算, 要先額外處理好條件狀況, 無法像 $\overbrace{\text{CSE}}$ 任意在參數內使用如 =MID() 函數等, 但似乎比較直接、比較好自學, 接著我們來介紹 =SUMIFS(), 之後再比較優缺點, 讓觀者自行決定要用哪個方式解題。

回到 9 道問題循序漸進來學習陣列運算,

1. 多少男性消費
2. 男性消費金額
3. 已婚男性消費金額
4. 已婚男性一年內消費金額
5. 49 歲已婚男性一年內消費金額
6. 20~30 歲已婚男性一年內消費金額
7. 20~30 及 40~50 歲男性一年內消費金額
8. 20~30 歲女性及 40~50 歲男性一年內消費金額
9. 20~30 歲女性及 40~50 歲未婚男性一年內消費金額

以上除第 1 題可以用 =COUNTIF() 解題 (見 7.3), 其餘 8 道題都可以用 =SUMIFS() 函數得到答案。

7.6.1 =SUMIFS()

=SUMIFS()

fx

會加總符合多項準則的所有引數。

```
= SUMIFS(sum_range, criteria_range1, criteria1,  
          [criteria_range2, criteria2, ...])
```

其中,

- sum_range: 要加總的儲存格範圍。
- criteria_range1: 使用 criteria1 測試的範圍。
- criteria1: 此準則定義要相加 criteria_range1 中的哪個儲存格。與 criteria1 組成一組搜尋配對, 用來在範圍中搜尋特定準則。找到範圍中的項目後, 即相加其 sum_range 中的對應值。
- criteria_range2: 第 2 組準則測試的範圍。
- criteria2: 與 criteria_range2 為第 2 組測試準則。

計算範圍內符合指定條件下指定要加總的儲存格範圍, 很繞口、很難懂的一串文字, 舉例來說, 「男性消費金額」: 男性就是指定條件, 消費金額就是要加總的範圍。再看, 「已婚男性消費金額」: 已婚為一組測試準則, 男性為另一組測試準則, 消費金額同上是加總範圍, 所以已婚男性就為 2 組測試準則。第 4 道題, 已婚男性一年內就為 3 組測試準則, 依此類推, 上列 9 道題都是幾組測試準則下要計算的消費金額, 所以 =SUMIFS() 就是在處理多測試準則下的加總問題, 但請注意, 這裡的多準則關係都要是交集, 對於第 7 道題後的交集性別、一年內外, 還有聯集年齡問題, 請見後面詳解。

7.6.2 資料設定

我們有 99 筆消費資料, 欄位有身分字號 (id)、年齡 (age)、婚姻 (married)、消費日 (consumer_date)、消費金額 (amount) 等 5 個變數, 與前一節相同的資料, 順使用 =SUMIFS() 來驗算一番。

我們的問題有性別, 所以在 B 欄插入一欄, 在新增的 B 欄中 [B1] 輸入

「sex」當變數名稱, 在 [B2] 格鍵入


= MID(A2,2,1)

複製到 [B3:B100],

	A	B	C	D	E	F
1	id	sex	age	married	consumer_date	amount
2	A169001035	1	63.67	1	2017/7/17	189437
3	B249356283	2	43.93	1	2017/6/2	328347
4	C151375012	1	49.00	1	2017/3/22	67319
5	D271532802	2	24.73	1	2016/4/15	99286
6	E154140527	1	2.27	0	2016/12/3	20348
7	F200381688	2	23.32	0	2016/12/26	42331
8	G110652243	1	39.28	0	2017/10/25	15892
9	H237507560	2	79.43	1	2017/7/1	71605
10	E150712936	1	55.91	0	2016/11/22	29241

產出性別欄, “1” 是男性, “2” 是女性, 現在資料新增為 6 個變數: 依序為 id, sex, age, married, consumer_date, amount。

選取 [A1:F100], Ctrl + Shift + F3 做變數命名 (見 7.1), 選「頂端列」為成稱, 分別命名為 id, sex, age, married, date, amount, 方便呼叫。



7.6.3 男性消費金額

男性消費金額, 用「男性」(sex=1) 當條件, 加總金額 (amount), 所以

=SUMIFS(amount, sex, 1)

其中意思為

- 第 1 個參數為要加總範圍, 是為 amount,
- 第 2 個參數條件一測試範圍, 性別測試範圍,
- 第 3 個參數為條件一的準則, 性別要是男性, 是為“1”。

	A	B	C	D	E	F	G	H	I
1	id	sex	age	married	consumer_date	amount			
2	A169001035	1	63.67	1	2017/7/17	189437			
3	B249356283	2	43.93	1	2017/6/2	328347		解答	
4	C151375012	1	49.00	1	2017/3/22	67319		51	#1 多少男性消費
5	D271532802	2	24.73	1	2016/4/15	99286		1052548	#2 男性消費金額
6	E154140527	1	2.27	0	2016/12/3	20348			#3 已婚男性消費金額

第 3 個參數為“1”或 1 都可被 =SUMIFS() 接受,

	A	B	C	D	E	F	G	H	I
1	id	sex	age	married	consumer_date	amount			
2	A169001035	1	63.67	1	2017/7/17	189437			
3	B249356283	2	43.93	1	2017/6/2	328347		解答	
4	C151375012	1	49.00	1	2017/3/22	67319		51	#1 多少男性消費
5	D271532802	2	24.73	1	2016/4/15	99286		1052548	#2 男性消費金額
6	E154140527	1	2.27	0	2016/12/3	20348			#3 已婚男性消費金額

另外, =SUMIFS() 的參數也可以輸入整欄, 如

=SUMIFS(F:F, B:B, "1")

	A	B	C	D	E	F	G	H	I
1	id	sex	age	married	consumer_date	amount			
2	A169001035	1	63.67	1	2017/7/17	189437			
3	B249356283	2	43.93	1	2017/6/2	328347		解答	
4	C151375012	1	49.00	1	2017/3/22	67319		51	#1 多少男性消費
5	D271532802	2	24.73	1	2016/4/15	99286		1052548	#2 男性消費金額
6	E154140527	1	2.27	0	2016/12/3	20348			#3 已婚男性消費金額

都得到一樣的答案。

7.6.4 已婚男性消費金額

已婚男性消費金額, 用「男性」(sex=1) 及「已婚」(married=1) 當條件, 加總金額 (amount), 所以

=SUMIFS(amount, sex, 1, married, 1)

	A	B	C	D	E	F	G	H	I
1	id	sex	age	married	consumer_date	amount			
2	A169001035	1	63.67	1	2017/7/17	189437			
3	B249356283	2	43.93	1	2017/6/2	328347		解答	
4	C151375012	1	49.00	1	2017/3/22	67319		51	#1 多少男性消費
5	D271532802	2	24.73	1	2016/4/15	99286		1052548	#2 男性消費金額
6	E154140527	1	2.27	0	2016/12/3	20348		489175	#3 已婚男性消費金額
7	F200381688	2	23.32	0	2016/12/26	42331			#4 已婚男性一年內消費金額

也可以為

=SUMIFS(F:F, B:B, 1, D:D, 1,)

都相同。

7.6.5 已婚男性一年內消費金額

已婚男性一年內消費金額,

- 男性: (sex=1),
- 已婚: (married=1) 當條件,
- 一年內: 則要化為 (date>"2016/11/5" 與 date<="2017/11/5") 當條件,

所以

=SUMIFS(amount, sex, 1, married, 1,
date, ">2016/11/5", date, "<=2017/11/5")

	C	D	E	F	G	H	I	J
	age	married	consumer_date	amount				
	63.67	1	2017/7/17	189437				
	43.93	1	2017/6/2	328347		解答		題目
	49.00	1	2017/3/22	67319		51	#1	多少男性消費
	24.73	1	2016/4/15	99286		1052548	#2	男性消費金額
	2.27	0	2016/12/3	20348		489175	#3	已婚男性消費金額
	23.32	0	2016/12/26	42331		388466	#4	已婚男性一年內消費金額 (2017/11/5)
	39.28	0	2017/10/25	15892			#5	49歲已婚男性一年內消費金額

也可以為

=SUMIFS(F:F, B:B, 1, D:D, 1, E:E, ">2016/11/5", E:E, "<=2017/11/5")

7.6.6 49 歲已婚男性一年內消費金額

49 歲已婚男性一年內消費金額,

- 男性: (sex=1) 條件一,
- 已婚: (married=1) 當條件二,
- 一年內: 以 (date>"2016/11/5" 與 date<="2017/11/5") 當條件三,
- 49 歲: (age=49) 為條件四。

所以

=SUMIFS(amount, sex, 1, married, 1,
date, ">2016/11/5", date, "<=2017/11/5", age, "=49")

fx		=SUMIFS(amount,sex,1,married,1,date,">2016/11/5",date,"<=2017/11/5",age,"=49")									
	D	E	F	G	H	I	J				
	married	consumer_date	amount								
7	1	2017/7/17	189437								
3	1	2017/6/2	328347		解答		題目				
0	1	2017/3/22	67319		51	#1	多少男性消費				
3	1	2016/4/15	99286		1052548	#2	男性消費金額				
7	0	2016/12/3	20348		489175	#3	已婚男性消費金額				
2	0	2016/12/26	42331		388466	#4	已婚男性一年內消費金額 (2017/11/5)				
8	0	2017/10/25	15892		67319	#5	49歲已婚男性一年內消費金額				
3	1	2017/7/1	71605			#6	20~30歲已婚男性一年內消費金額				

也可以為

=SUMIFS(F:F, B:B, 1, D:D, 1,
E:E, ">2016/11/5", E:E, "<=2017/11/5", C:C, "=49")

7.6.7 20~30 歲已婚男性一年內消費金額

20~30 歲已婚男性一年內消費金額,

- 男性: (sex=1) 條件一,
- 已婚: (married=1) 當條件二,

- 一年內: 以 (date>"2016/11/5" 與 date<="2017/11/5") 當條件三,
- 20~30 歲: (age>=20 與 age<30) 為條件四。

所以

=SUMIFS(amount, sex, 1, married, 1,
date, ">2016/11/5", date, "<=2017/11/5",
age, ">=20", age, "<30")

fx		=SUMIFS(amount,sex,1,married,1,date,">2016/11/5",date,"<=2017/11/5",age,">=20",age,"<30")									
	D	E	F	G	H	I	J				
	married	consumer_date	amount								
7	1	2017/7/17	189437								
8	1	2017/6/2	328347		解答					題目	
9	1	2017/3/22	67319		51	#1	多少男性消費				
10	1	2016/4/15	99286		1052548	#2	男性消費金額				
11	0	2016/12/3	20348		489175	#3	已婚男性消費金額				
12	0	2016/12/26	42331		388466	#4	已婚男性一年內消費金額 (2017/11/5)				
13	0	2017/10/25	15892		67319	#5	49歲已婚男性一年內消費金額				
14	1	2017/7/1	71605		56974	#6	20~30歲已婚男性一年內消費金額				
15	0	2016/11/22	29241			#7	20~30及40~50歲男性一年內消費金額				

也可以為

=SUMIFS(F:F, B:B, 1, D:D, 1, E:E, ">2016/11/5", E:E, "<=2017/11/5",
C:C, ">=20", C:C, "<30")

7.6.8 20~30 及 40~50 歲男性一年內消費金額

20~30 及 40~50 歲男性一年內消費金額, 用年齡分二塊,

- 第一部分: 20~30 歲,
 - 男性: (sex=1) 條件一,
 - 一年內: 以 (date>"2016/11/5" 與 date<="2017/11/5") 當條件二,
 - 20~30 歲: (age>=20 與 age<30) 為條件三。
- 第二部分: 40~50 歲,
 - 男性: (sex=1) 條件一,
 - 一年內: 以 (date>"2016/11/5" 與 date<="2017/11/5") 當條件二,
 - 40~50 歲: (age>=40 與 age<50) 為條件三。

所以

$$\begin{aligned}
 &= \text{SUMIFS}(\text{amount}, \text{sex}, 1, \text{date}, ">2016/11/5", \text{date}, "<=2017/11/5", \\
 &\quad \text{age}, ">=20", \text{age}, "<30") \\
 &+ \text{SUMIFS}(\text{amount}, \text{sex}, 1, \text{date}, ">2016/11/5", \text{date}, "<=2017/11/5", \\
 &\quad \text{age}, ">=40", \text{age}, "<50")
 \end{aligned}$$

fx		=SUMIFS(amount,sex,1,date,">2016/11/5",date,"<=2017/11/5",age,">=20",age,"<30")+SUMIFS(amount,sex,1,date,">2016/11/5",date,"<=2017/11/5",age,">=40",age,"<50")									
D	E	F	G	H	I	J					
married	consumer_date	amount									
1	2017/7/17	189437									
1	2017/6/2	328347		解答		題目					
1	2017/3/22	67319		51	#1	多少男性消費					
1	2016/4/15	99286		1052548	#2	男性消費金額					
0	2016/12/3	20348		489175	#3	已婚男性消費金額					
0	2016/12/26	42331		388466	#4	已婚男性一年內消費金額 (2017/1/5)					
0	2017/10/25	15892		67319	#5	49歲已婚男性一年內消費金額					
1	2017/7/1	71605		56974	#6	20~30歲已婚男性一年內消費金額					
0	2016/11/22	29241		161993	#7	20~30及40~50歲男性一年內消費金額					
1	2017/10/10	68618			#8	20~30歲女性及40~50歲男性一年內消費金額					

也可以為

$$\begin{aligned}
 &= \text{SUMIFS}(\text{F:F}, \text{B:B}, 1, \text{E:E}, ">2016/11/5", \text{E:E}, "<=2017/11/5", \\
 &\quad \text{C:C}, ">=20", \text{C:C}, "<30") \\
 &+ \text{SUMIFS}(\text{F:F}, \text{B:B}, 1, \text{E:E}, ">2016/11/5", \text{E:E}, "<=2017/11/5", \\
 &\quad \text{C:C}, ">=40", \text{C:C}, "<50")
 \end{aligned}$$

7.6.9 20~30 歲女性及 40~50 歲男性一年內消費金額

20~30 歲女性及 40~50 歲男性一年內消費金額，一樣分成二部分，

- 第一部分: 20~30 歲女性,
 - 女性: (sex=2) 條件一,
 - 一年內: 以 (date>"2016/11/5" 與 date<="2017/11/5") 當條件二,
 - 20~30 歲: (age>=20 與 age<30) 為條件三。
- 第一部分: 40~50 歲男性,

- 男性: (sex=1) 條件一,
- 一年內: 以 (date>"2016/11/5" 與 date<="2017/11/5") 當條件二,
- 40~50 歲: (age>=40 與 age<50) 為條件三。

所以

$$=SUMIFS(amount, sex, 2, date, ">2016/11/5", date, "<=2017/11/5", age, ">=20", age, "<30")$$

$$+ SUMIFS(amount, sex, 1, date, ">2016/11/5", date, "<=2017/11/5", age, ">=40", age, "<50")$$

D	E	F	G	H	I	J
married	consumer_date	amount				
1	2017/7/17	189437				
1	2017/6/2	328347				
1	2017/3/22	67319		51	#1	多少男性消費
1	2016/4/15	99286		1052548	#2	男性消費金額
0	2016/12/3	20348		489175	#3	已婚男性消費金額
0	2016/12/26	42331		388466	#4	已婚男性一年內消費金額 (2017/11/5)
0	2017/10/25	15892		67319	#5	49歲已婚男性一年內消費金額
1	2017/7/1	71605		56974	#6	20~30歲已婚男性一年內消費金額
0	2016/11/22	29241		161993	#7	20~30及40~50歲男性一年內消費金額
1	2017/10/10	68618		528797	#8	20~30歲女性及40~50歲男性一年內消費金額
0	2016/6/8	65509			#9	20~30歲女性及40~50歲未婚男性一年內消費金額
1	2017/10/11	67008				

也可以為

$$=SUMIFS(F:F, B:B, 2, E:E, ">2016/11/5", E:E, "<=2017/11/5", C:C, ">=20", C:C, "<30")$$

$$+ SUMIFS(F:F, B:B, 1, E:E, ">2016/11/5", E:E, "<=2017/11/5", C:C, ">=40", C:C, "<50")$$

7.6.10 20~30 歲女性及 40~50 歲未婚男性一年內消費金額

20~30 歲女性及 40~50 歲未婚男性一年內消費金額, 同前分成二部分,

- 第一部分: 20~30 歲女性,
 - 女性: (sex=2) 條件一,

- 一年內: 以 (date>"2016/11/5" 與 date<="2017/11/5") 當條件二,
 - 20~30 歲: (age>=20 與 age<30) 為條件三。
- 第一部分: 40~50 歲未婚男性,
 - 男性: (sex=1) 條件一,
 - 一年內: 以 (date>"2016/11/5" 與 date<="2017/11/5") 當條件二,
 - 40~50 歲: (age>=40 與 age<50) 為條件三,
 - 未婚: (married=0) 為條件四。

所以

$$\begin{aligned}
 &= \text{SUMIFS}(\text{amount}, \text{sex}, 2, \text{date}, ">2016/11/5", \text{date}, "<=2017/11/5", \\
 &\quad \text{age}, ">=20", \text{age}, "<30") \\
 &+ \text{SUMIFS}(\text{amount}, \text{sex}, 1, \text{date}, ">2016/11/5", \text{date}, "<=2017/11/5", \\
 &\quad \text{age}, ">=40", \text{age}, "<50", \text{married}, "=0")
 \end{aligned}$$

D	E	F	G	H	I	J
married	consumer_date	amount				
1	2017/7/17	189437				
1	2017/6/2	328347				
1	2017/3/22	67319		51	#1	多少男性消費
1	2016/4/15	99286		1052548	#2	男性消費金額
0	2016/12/3	20348		489175	#3	已婚男性消費金額
0	2016/12/26	42331		388466	#4	已婚男性一年內消費金額 (2017/11/5)
0	2017/10/25	15892		67319	#5	49歲已婚男性一年內消費金額
1	2017/7/1	71605		56974	#6	20~30歲已婚男性一年內消費金額
0	2016/11/22	29241		161993	#7	20~30及40~50歲男性一年內消費金額
1	2017/10/10	68618		528797	#8	20~30歲女性及40~50歲男性一年內消費金額
0	2016/6/8	65509		455306	#9	20~30歲女性及40~50歲未婚男性一年內消費金額
1	2017/10/11	67008				

也可以為

$$\begin{aligned}
 &= \text{SUMIFS}(\text{F:F}, \text{B:B}, 2, \text{E:E}, ">2016/11/5", \text{E:E}, "<=2017/11/5", \\
 &\quad \text{C:C}, ">=20", \text{C:C}, "<30") \\
 &+ \text{SUMIFS}(\text{F:F}, \text{B:B}, 1, \text{E:E}, ">2016/11/5", \text{E:E}, "<=2017/11/5", \\
 &\quad \text{C:C}, ">=40", \text{C:C}, "<50", \text{D:D}, "=0")
 \end{aligned}$$

7.6.11 {=SUM(IF())} 與 =SUMIFS() 比較

學到這裡,有人會認為 =SUMIFS() 比較直覺、比較簡單、比較好用,的確, =SUMIFS() 不需要先有陣列基礎知識,不需要知道 **Ctrl** + **Shift** + **Enter**,但當我們把這些方法當列在一起時,就可以發現它的優缺點。

項目	{=SUM(IF())}	=SUMIFS()
運算	Ctrl + Shift + Enter	Enter
原理	陣列運算	Excel 內建函數
功能	每個參數皆能使用函數運算	資料無法在參數內運算
優勢	功能強大,可以任意變化求解過程	易懂易學,尤其是自學者
缺點	要有陣列運算基本能力,門檻高	單純,變化較少

如 7.5 所說明, {=SUM(IF())} 的 SUM 是可以換的,雖 Excel 內建函數除了 =SUMIFS() 還有 =COUNTIFS()、=AVERAGE() 等,但陣列函數還是比較寬廣,譬如: 20~30 歲已婚男性一年內消費第 5 大金額?

$$\{=LARGE(IF((MID(id, 2, 1)="1")*(married=1) * ("2017/11/5"-date<365)*((age>=20)*(age<30)), amount), 5)\}$$

若只是存粹要把題目順利求解,只要自己隨手能求得解答就是好方法,

7.7 表格

以上演示都是單格,其實 Excel 的強大在於拖曳複製,所以善用 F4 鍵控制住參數的絕對位址就能輕鬆地製表。

很多人喜歡用樞紐分析表製表,樞紐分析製造出來的表格有 2 大缺失,

1. 不會自動更新
2. 縱、橫坐標依電腦對於變數的排序,而且自己忽略 o 的變數

陣列運算會即時更新數字,只要資料表更新,它就更新。另外樞紐自行製造出來的表格有可能因為資料不同而產生不同大小位置的表格,這點很困擾,所以我不喜歡。

第 8 章

陣列運算的運用

8.1 用函數排序

Excel 的排序功能在工具列上, 有「排序與篩選」按鈕, 按下之後選擇你要的排序方式就可達到, 但若要不影響原來陣列, 用公式創造出來一欄新的陣列而且是經過排序的, 請看以下介紹。

若要排序 A1:A5 這 5 個數字到 B1:B5, 首先記得選取 B1:B5 儲存格, 然後輸入

$$\{= \text{LARGE}(A1:A5, \{1; 2; 3; 4; 5\})\}$$

這是數值由大排到小, 若要反過來由數字小排到大, 可以把第二個參數 {1; 2; 3; 4; 5} 改成 {5; 4; 3; 2; 1}, 又或者是使用 SMALL() 函數,

$$\{= \text{SMALL}(A1:A5, \{1; 2; 3; 4; 5\})\}$$

當你的資料數很大時, 第 2 個參數 {1; 2; 3; 4; 5} 可以改以參照儲存格, 如: C1:C999 等, 而 C1:C999 只要輸入 1 ~ 999 即可。

進一步來說, 參照儲存格可以改成 ROW() 函數, 免去手動製造一欄 1 ~ 999 數列。一般 ROW() 函數會傳回該參數中儲存格的「列號」, 如 ROW(D32)=32, 若不輸入參數則回傳 ROW() 函數所在列號。結合陣列運算特性, 當我們選取 E6:E10, 輸入 {=ROW(E1:E5)} 時, 會回傳 {1; 2; 3; 4; 5}, 意思是回傳其中 5 個元素各自的列號; 又對 ROW() 函數來說, 「欄」根本不重要, 直接大器地省略, 就變成 {=ROW(1:5)}, 一樣會回傳 {1; 2; 3; 4; 5}。所以運用此特性, 在不用參照儲存格下, 我們可以直接把算式下成:

$$\{= \text{SMALL}(A1:A5, \text{ROW}(1 : 999))\}$$

ROW() 函數可以產生一欄的整數陣列,我們可以把它應用到 COLUMN() 函數,也就是說 COLUMN() 函數可以產生一列整數陣列,如 {=COLUMN(A6:C6)} 產生 {1,2,3}, 參數部分改成欄號 {=COLUMN(A:C)} 會產生 {1,2,3}, 最後用數字的欄號 {=COLUMN(1:3)} 一樣會產生 {1,2,3}。

8.2 用函數篩選

上節介紹排序,這節來介紹篩選,篩選會比排序更難點。

回到前一章的消費資料,假設我們要篩選出 id 以 A 開頭的資料。

找出前一章資料,從 A 欄到 E 欄分別是 age, amount, date, id, married 等 5 個變數,我們先在 F1 欄加入 num 欄位名稱,並自 F2 到 F100 填入 2 ~ 100。接著命名,選取 A1 到 F100,用 Ctrl + Shift + F3 三鍵定義名稱的程序,選取「頂端列」,將欄位變數命好名,點選 Excel 工具區的

公式 | 定義名稱

將 A1:E100 定義為“data”。

先在 G1 儲存格輸入“A”,這個 A 是我們要篩選 id 開頭的 A。然後在 H2 欄輸入

$$\{= \text{IFERROR}(\text{INDEX}(\text{data}, \text{SMALL}(\text{IF}(\text{LEFT}(\text{id}, 1) = \text{\$G\$1}, \text{num}), \text{ROW}() - 1), 1), \text{""})\}$$

在 I2 欄輸入

$$\{= \text{IFERROR}(\text{INDEX}(\text{data}, \text{SMALL}(\text{IF}(\text{LEFT}(\text{id}, 1) = \text{\$G\$1}, \text{num}), \text{ROW}() - 1), 2), \text{""})\}$$

在 J2 欄輸入

$$\{= \text{IFERROR}(\text{INDEX}(\text{data}, \text{SMALL}(\text{IF}(\text{LEFT}(\text{id}, 1) = \text{\$G\$1}, \text{num}), \text{ROW}() - 1), 3), \text{""})\}$$

在 K2 欄輸入

$$\{= \text{IFERROR}(\text{INDEX}(\text{data}, \text{SMALL}(\text{IF}(\text{LEFT}(\text{id}, 1) = \text{\$G\$1}, \text{num}), \text{ROW}() - 1), 4), \text{""})\}$$

在 L2 欄輸入

$$\{= \text{IFERROR}(\text{INDEX}(\text{data}, \text{SMALL}(\text{IF}(\text{LEFT}(\text{id}, 1) = \text{\$G\$1}, \text{num}), \text{ROW}() - 1), 5), "")\}$$

選取 H2 到 L2, 複製到 H3:L100, 大功告成。如果你有把握你要篩選的資料為 30 列, 當然你可以只複製到 30 列即可, 不必複製到 100 列; 這邊的作法是複製到 100 列, 若是沒有資料的列, 最後也會被 IFERROR() 函數改為空白。

當我們把 G1 儲存格改成 “B” 時, H2:L100 會自動篩選出以 B 為 id 開頭的資料, 試試看, 自由更改為 C、D、E、...

8.2.1 =SMALL()

見 ?? =SMALL() 函數有 2 個參數,

fx

=SMALL()

傳回第一個參數資料集中的第 k 小的值, 邏輯值及文字將被略過而不計。若參數中沒有數字, 則產生錯誤。

= SMALL(array, k)

其中,

- array: 要判斷第 k 個最大值得陣列或資料範圍。
- k : 要傳回之資料陣列或儲存格範圍中的位置 (由最小起算)。

回到篩選的主題, 我們取 H2 儲存格公式來剖析, 由內往外看。首先,

$$\{= \text{IF}(\text{LEFT}(\text{id}, 1) = \text{\$G\$1}, \text{num})\}$$

IF() 函數有 3 個參數, 第 3 個參數 (若為 FALSE) 回傳 FALSE, 反正它無足輕重, 所以我把它省略。所以這個式子的邏輯是, 若 id 開頭是 G1 儲存格, 回傳我們創造出的 num (編號), 而這個編號正是它的列號。

所以上式的答案是

$$\{2; \text{FALSE}; \text{FALSE}; \text{FALSE}; \text{FALSE}; \text{FALSE}; \text{FALSE}; \text{FALSE}; \dots\}$$

然後我們還要加上 SMALL() 函數。

把我們的答案代入目標式子,變成

$$\{= \text{SMALL}(\{2; \text{FALSE}; \dots\}, \text{ROW}()-1)\}$$

ROW() 函數回傳該列號,因為我們在 H2 儲存格輸入此公式,所以 ROW()-1 = 2 - 1 = 1, 於是上式就是在 {2; FALSE;...} 中求第 1 小的值。H3 儲存格就是在求第 2 小的值, 依此類推, H100 儲存格就是在求第 99 小的值。

回到 H2 儲存格, 找第 1 小的值自然是 2, 這個“2”是指我們創造出來的那 num 變數, 那個變數是取列號, 所以 2 是指第 2 列的那個 id; 同理, H3 格找第 2 小的值自然是 11, 11 是第 11 列的那個 id。

因為資料中只有 10 個 id 是以 A 開頭, 所以 H12 儲存格求第 11 小的值不存在, 故 SMALL() 函數回傳錯誤的“#NUM”, 當然 H13 以下亦是。

陣列運算其實到此 SMALL() 已經結束, 因為此例中的 SMALL() 回傳出的結果已經是一個單值, 不再是一個一維以上陣列, 所以更外圍的 INDEX() 與 IFERROR() 都回歸到一維的計算。附帶一說, $\overbrace{\text{CSE}}$ 用在一維度計算時, 與  鍵無異, 所以 Excel 還是照常運算。

8.2.2 =INDEX()

=INDEX()

fx

=INDEX() 函數有 3 個參數, 傳回數值或是位於給定範圍內特定列或欄交集處的儲存格參照。

$$= \text{INDEX}(\text{array}, \text{row_num}, \text{column_num})$$

其中,

- array: 資料或陣列範圍。
- row_num: 會選取陣列中傳回值的列。
- column_num: 會選取陣列中傳回值的欄。

譬如說, INDEX(E3:K100, 14, 6) 會回傳 J16 儲存格的值, 因為自 E3 儲存格, 往下 14 列, 往右 6 欄, 所以是 J16 儲存格。

剛才解到 H2 儲存格的 {SMALL(IF(LEFT(id, 1)=\$G\$1, num), ROW()-1)}=2,

所以

$$\begin{aligned} & \{ = \text{INDEX}(\text{data}, \text{SMALL}(\text{IF}(\text{LEFT}(\text{id}, 1) = \$G\$1, \text{num}), \text{ROW}() - 1), 1) \} \\ & = \text{INDEX}(\text{data}, 2, 1) \end{aligned}$$

就是指 A2 儲存格, 等於 “A169001035”。

H3 儲存格就指 INDEX(data, 11, 1), 為 A11 儲存格, 回傳 “A272535343”; 依此類推, 直到 H11 儲存格的錯誤,

$$\{ = \text{INDEX}(\text{data}, \#\text{NUM}, 1) \}$$

回傳還是 #NUM。

8.2.3 =IFERROR()

=IFERROR()

fx

IFERROR() 函數就是在處理 INDEX() 函數的錯誤回傳, 如果第 1 個參數有錯誤, 回傳第 2 個參數, 當第 1 個參數沒有錯誤, 回傳第 1 個參數。

$$= \text{IFERROR}(\text{value}, \text{value_if_error})$$

其中,

- value: 被檢查是否有錯誤的輸入。
- value_if_error: 當錯誤時回傳的值。

只要我們設定 =IFERROR() 函數的第 2 個參數為空集合 “”, 一旦有錯誤如 #NUM 進入第 1 個參數後, 函數自然會回傳第 2 個參數的參照或值, 就為空白, 所以它可以處理上小節 INDEX() 函數所產生的錯誤。

$$= \text{IFERROR}(\text{A272535343}, "")$$

第 1 個參數為普通的字串, 所以沒有錯誤, 它回傳還是 A272535343。若如 H12 儲存格的錯誤的話,

$$= \text{IFERROR}(\#\text{NUM}, "")$$

回傳就會是空白。

做這件事的好處是, 當我們把 G1 儲存格改成 “B” 時, 即要篩選出 id 以 B 開頭時, 它不一定會與 A 開頭數量一樣, 所以它可以保持篩選出來的資料乾淨整潔, 不受 #NUM 影響, 能繼續往下運算。

8.3 抽樣—抽出不放回

若我們要在 {1, 2, 3, ..., 15}, 共 15 個數字中抽取一個整數, 假設每個數字抽中的機率都要一樣, 我們可以用 =RAND() 函數。

進入抽出不放回前, 我們先看抽出放回的情形。

抽出放回

見 ?? =RAND() 函數,

=RAND()

fx

傳回大於或等於 0 且小於等於 1 的均等分配隨機實數。每次計算工作表時, 都會傳回新的隨機實數。

= RAND()

其中並不需要引數、參數。

=RAND() 函數傳回大於等於 0 且小於 1 的均等分配隨機實數。每次計算工作表時, 都會傳回新的隨機實數。操作時不用輸入參數, 但要輸入括號, 括號內不用有參數, 自然產生介於 0~1 的小數。利用這個特性, 我們想要取一個介於 $a \sim b$ 間的實數, 只要

$$= \text{RAND}() * (b - a) + a$$


以上你會得到一個「實數」, 帶小數的實數, 若我們要的是一個「整數」, 只要加入一個整數函數稍做修改,

$$= \text{INT}(\text{RAND}() * (b - a) + a)$$

見 ??, =INT() 函數是取整數, 小數部分無條件捨去。回到我們的需求, 因為我們要整數, 所以把小數部分捨去, 就是我們所要的了。


若是在 $[1, 15]$ 之間取一個數，則

$$= \text{RAND}() * (15 - 1) + 1$$



取出 10.88949，它是帶小數的一個數。若是我們要在 $\{1, 2, 3, \dots, 15\}$ 的整數間取一個整數，則

$$= \text{INT}(\text{RAND}() * (15 - 1) + 1)$$



取出的數字就為 2。若是你再做一次，則還是有可能抽到 2，這就是抽出放回的概念。而接著我們要介紹的就是抽出一個數字後，不會再重複抽到同樣數字。

抽出不放回

爲了要讓我們容易理解，我們要在 $\{1, 2, 3, \dots, 10\}$ 間，分 10 次抽樣，不偏不倚，每個數字僅抽出一個，而且不會重複，這就是抽出不放回。

假設我們在 A2 到 A11 儲存格把這 10 個數字抽出，則在 A2 格輸入

$$\{=\text{LARGE}(\text{IF}(\text{COUNTIF}(\$A\$1:A1, \text{ROW}(\$1:\$10)) = 0, \text{ROW}(\$1:\$10)), \text{ROUNDUP}(\text{RAND}() * (11 - \text{ROW}(A1)), 0))\}$$

將之複製到 A3 至 A11，則可在其中分別得到 $\{1, 2, 3, \dots, 10\}$ 各一次的數字。

對於複雜的上式，我們分開來介紹講解其中結構與運作方式，從內到外循序漸進分爲：

1. {=ROW()}
2. {=COUNTIF(., Row())}
3. {=IF(COUNTIF())}
4. {=LARGE(IF(COUNTIF()))}

8.3.1 {=ROW()}

見 ??, =ROW() 函數傳回的是指定儲存格的列號, 當其中的參數不輸入時, 傳回的是該函數所在的列號。


以上是前面介紹過的, 現在, 當函數外面加了大括號, 就會變成陣列, 在 A2 儲存格輸入

{= ROW(\$1 : \$10)}


或

{= ROW(\$A\$1 : \$A\$10)}

都代表 A2 儲存格為 {1; 2; 3; ...; 10} 的陣列, 但外觀上, A2 儲存格只會顯現第一個數字“1”。



按 *fx* 鍵看內容,



就可知道實際內容為 {1; 2; 3; 4; 5; 6; 7; 8; 9; 10}, 當我把它複製到 A2 至 A11, 每個儲存格只會顯示“1”, 但實際上每個儲存格都是一個獨立的 {1; 2; 3; ...; 10} 的陣列。

	A	B	C	D	E	F
1						
2	1					
3	1					
4	1					
5	1					
6	1					
7	1					
8	1					
9	1					
10	1					
11	1					

我們把這個陣列放到 =COUNTIF() 函數中。

8.3.2 {=COUNTIF()}

接著我們看 =COUNTIF() 的內容。

$$\{= \text{COUNTIF}(\$A\$1 : \$A1, \text{ROW}(\$1 : \$10))\}$$

=COUNTIF() 函數可見 ??, 其為計算範圍內符合指定條件儲存格的數目。

請注意上式的第一個參數 \$A\$1 : \$A1 是浮動的條件, 當它往下複製, 複製到 A10 格時, 該參數跟著變為 \$A\$1 : \$A10。

當 A2 格輸入上式計算式時, 其實是在比對 A1 格內有多少數字為 {1; 2; 3; ...; 10}, 因為 A1 格為空白, 所以答案會是 {0; 0; ...; 0}。接著把公式複製到 A3 儲存格, 意思是在計算 A1:A2 多少數字為 {1; 2; 3; ...; 10}, 在此, 因為 A1 為空白, A2 為 0, 所以答案自然還是 {0; 0; ...; 0}。這邊的 =COUNTIF() 主要是用來計算有多少數字被抽出來過。

要清楚看此函數的功用, 我們先在 A2:A11 抽出不放回抽出 10 個數字, 如下圖,

	A	B	C	D	E	F	G	H	I	J	K
1	1										
2	6										
3	5										
4	4										
5	10										
6	2										
7	7										
8	9										
9	3										
10	8										
11	1										

然後選取 A14:A23 儲存格, 輸入

$$\{= \text{COUNTIF}(A\$1:A1),\text{ROW}(\$1:\$10)\}$$

B14:B23 輸入

$$\{= \text{COUNTIF}(A\$1:A2),\text{ROW}(\$1:\$10)\}$$

依此, 分別輸入到 J14:J23 為

$$\{= \text{COUNTIF}(A\$1:A10),\text{ROW}(\$1:\$10)\}$$

如下圖,

	A	B	C	D	E	F	G	H	I	J
13	1	2	3	4	5	6	7	8	9	10
14	0	0	0	0	0	0	0	0	0	0
15	0	0	0	0	0	0	1	1	1	1
16	0	0	0	0	0	0	0	0	0	1
17	0	0	0	1	1	1	1	1	1	1
18	0	0	1	1	1	1	1	1	1	1
19	0	1	1	1	1	1	1	1	1	1
20	0	0	0	0	0	0	1	1	1	1
21	0	0	0	0	0	0	0	0	0	1
22	0	0	0	0	0	0	0	1	1	1
23	0	0	0	0	1	1	1	1	1	1

這裡的 A14:A23 就是上述計算 A1 格有多少數字為 {1; 2; 3; ...; 10}, 答案正是 {0; 0; ...; 0}, 主要因為 A1 格為空白。接著, A2 格為 6, B14:B23 計算 A1:A2

格有多少數字為 {1; 2; 3; ...; 10}, 答案就會變成 {0; 0; 0; 0; 0; 1; 0; 0; 0; 0}, 在 6 的地方變為 1; 同理 A₃ 格為 5, C₁₄:C₂₃ 計算的結果為 {0; 0; 0; 0; 0; 1; 1; 0; 0; 0; 0}, 直到 J₁₄:J₂₃, 2 至 10 的數字都被抽出了, 只剩下 1, 所以 {0; 1; 1; 1; 1; 1; 1; 1; 1; 1}。

{=COUNTIF()} 在整個算式中扮演記錄抽出與未抽出的分類, 這個功能結合以下 {=IF()} 函數, 來標籤抽出與未抽出, 抽出的數字, 就貼上 FALSE 標籤, 並且自抽獎箱抽離, 未抽出的數字就仍留在抽獎箱中等待抽出。

8.3.3 {=IF(COUNTIF())}

這邊的問題用白話講就會是, 「如果以上未出現過的數字, 就回傳各自原應該是 1 到 10 的數字, 否則就回傳 FALSE。」換句話說,

$$\{=IF(COUNTIF(\$A\$1:A1, ROW(\$1:\$10)) = 0, ROW(\$1:\$10), FALSE)\}$$

可以把第 3 個參數 value_false 省略所以就變成

$$\{=IF(COUNTIF(\$A\$1:A1, ROW(\$1:\$10)) = 0, ROW(\$1:\$10))\}$$

回到剛才 A₁₄:A₂₃ 為 {0; 0; ...; 0}, {=IF()} 會回傳 {1; 2; ...; 10}; 若 B₁₄:B₂₃ 時, {=IF()} 會回傳 {1; 2; 3; 4; 5; FALSE; 7; 8; 9; 10}; 依此類推, J₁₄:J₂₃ 時, {=IF()} 的答案會變成 {1; FALSE; FALSE; FALSE; FALSE; FALSE; FALSE; FALSE; FALSE; FALSE; FALSE};

	A	B	C	D	E	F	G	H	I	J	K
25	1	1	1	1	1	1	1	1	1	1	
26	2	2	2	2	2	FALSE	FALSE	FALSE	FALSE	FALSE	
27	3	3	3	3	3	3	3	3	3	FALSE	
28	4	4	4	FALSE							
29	5	5	FALSE								
30	6	FALSE									
31	7	7	7	7	7	7	FALSE	FALSE	FALSE	FALSE	
32	8	8	8	8	8	8	8	8	8	FALSE	
33	9	9	9	9	9	9	9	9	FALSE	FALSE	
34	10	10	10	10	FALSE	FALSE	FALSE	FALSE	FALSE	FALSE	

8.3.4 {=LARGE(IF(COUNTIF()))}

=LARGE() 傳回資料範圍 (第 1 個參數) 中第 k (第 2 個參數) 大的值。

我們把 {=IF()} 所產生的陣列資料丟到 =LARGE() 函數中當第 1 個參數，就會變成被挑選的資料範圍，再從中隨機取出第 k 大的值。

$$\{= \text{LARGE}(\text{IF}(\text{COUNTIF}(\$A\$1:A1, \text{ROW}(\$1:\$10)) = 0, \text{ROW}(\$1:\$10)), \text{ROUNDUP}(\text{RAND}() * (11 - \text{ROW}(A1)), 0))\}$$

把 =LARGE() 改成 =SMALL 也是一種方式，

$$\{= \text{SMALL}(\text{IF}(\text{COUNTIF}(\$A\$1:A1, \text{ROW}(\$1:\$10)) = 0, \text{ROW}(\$1:\$10)), \text{ROUNDUP}(\text{RAND}() * (11 - \text{ROW}(A1)), 0))\}$$

又或者把 =COUNTIF 製換成 =SUMIF 也一樣，

$$\{= \text{LARGE}(\text{IF}(\text{SUMIF}(\$A\$1:A1, \text{ROW}(\$1:\$10)) = 0, \text{ROW}(\$1:\$10)), \text{ROUNDUP}(\text{RAND}() * (11 - \text{ROW}(A1)), 0))\}$$

第 9 章

例子演練

9.1 半年內消費筆數

假設有一共 800 筆的消費資料, A 欄為客戶 id, B 欄為消費日期, C 欄為消費金額組成的 $\{800 \times 3\}$ 陣列。

A10031386	13Oct2015	856
A10031386	20Nov2015	742
A10031386	16Dec2015	600
A10031386	19Jan2016	1210
A10031386	17Jun2016	35
A10031395	19Jan2016	798
A10031484	15Jan2016	665
A10031484	12Apr2016	89
⋮	⋮	⋮
Z20052143	02Jan2016	226

要計算的問題是「每個 id 的第一筆消費後的半年內共消費多少筆」, 所以半年期間會隨每個客戶不同, 並非是統一一致的某日起半年內。

解法:

1. 將 A:B 欄 (id, 消費日期) 資料複製至 G1 格;
2. 分別依 id, 消費日期等 2 個分別為主要及次要層級排序, 消費日期為遞增;
3. 排序完後執行 excel 的 資料 | 移除重複 功能, 以 id 刪除重複值, 如此一來只會剩下唯一 id 及該客戶第一筆消費。
4. 在 I1 格輸入 $\{=SUM(IF((\$A\$1:\$A\$800=G1)* (B1-H1<365/2),1,0))\}$

當 A 欄等於 G 欄, 原始資料客戶等於我們想要統計的客戶時, 同時該名客戶消費日落於 H 欄第一筆消費半年內的就為我們要統計的次數。